

Notes of IDL Programming V1.0

Faxiang Zheng

July 14, 2016

1 Before you start

Assuming that you have IDL (version greater than or equal to 7.1) installed on your Linux, MacOS or FreeBSD, there's some additional configuration that I recommend.

1.1 install Coyote and Catalyst Library

Open your terminal (with bash or zsh), then copy and paste following commands

```

1 mkdir -p ~/lib/idl/idl-coyote # ~/lib/idl can be anywhere else
2 cd ~/lib/idl/idl-coyote
3 git init
4 git remote add origin https://github.com/davidwfanning/idl-coyote.git
5 echo "coyote" >> .git/info/sparse-checkout
6 echo "catalyst" >> .git/info/sparse-checkout
7 git config core.sparsecheckout true
8 git pull origin master

```

1.2 install AstroLib

```

1 cd ~/lib/idl
2 git clone https://github.com/wlandsman/IDLAstro.git astrolib

```

1.3 install Motley Library

```

1 cd ~/Downloads
2 wget http://www.idlcoyote.com/hadfield/idl/motley.tar.gz
3 mkdir ~/lib/idl/motley
4 tar -C ~/lib/idl/motley -zvxf motley.tar.gz

```

1.4 install Gumley's IDL routines

```

1 cd ~/Downloads
2 wget https://svn.ssec.wisc.edu/repos/geoffc/IDL/Gumley/PIP_programs.tar.Z
3 mkdir ~lib/idl/gumley
4 tar -C ~lib/idl/gumley -zvxf ~/Downloads/PIP_programs.tar.Z

```

1.5 install Slug's IDL routines

```

1 cd ~/Downloads
2 wget http://slugidl.pbworks.com/f/slug_idl.tar
3 mkdir ~lib/idl/slug
4 tar -C ~lib/idl/slug -xvf slug_idl.tar

```

1.6 install TeXtoIDL

```

1 cd ~/Downloads
2 wget http://physics.mnstate.edu/craig/textoidl/textoidl-2-1-2.tar
3 tar -C ~lib/idl -xvf textoidl-2-1-2.tar

```

1.7 setting PATH

Add the following to your `~/.profile` (bash user) or `~/.zshenv` (zsh user)

```

1 # IDL setting
2 export IDL_LIB_DIR=$HOME/lib/idl
3 export IDL_PATH='<IDL_DEFAULT>'
4 export IDL_PATH=$IDL_PATH:+'$IDL_LIB_DIR/idl-coyote/coyote'
5 export IDL_PATH=$IDL_PATH:+'$IDL_LIB_DIR/idl-coyote/catalyst'
6 export IDL_PATH=$IDL_PATH:+'$IDL_LIB_DIR/astrolib/pro'
7 export IDL_PATH=$IDL_PATH:+'$IDL_LIB_DIR/motley'
8 export IDL_PATH=$IDL_PATH:+'$IDL_LIB_DIR/gumley'
9 export IDL_PATH=$IDL_PATH:+'$IDL_LIB_DIR/slug'
10 export IDL_PATH=$IDL_PATH:+'$IDL_LIB_DIR/textoidl'
```

1.8 startup file

You can, of course, name the start-up file anything you like and locate it anywhere in your directory structure. For example, add the following to your `~/.profile` (bash user) or `~/.zshenv` (zsh user)

```
1 export IDL_STARTUP=$HOME/.idlstartup.pro
```

I put something like

```
1 print, 'Hello, ' + getenv('USER') + '! ' + 'IDL is ready :-)'
```

in my startup file.

1.9 editor

Emacs, with its embedded *IDLWave-mode* and *IDLWave-shell*, is a little slice of heaven for dealing with IDL source file and command line. You may have a look at my Emacs configuration.

If your IDL version is 7.1, you might have to do something to fix a bug.

- open your terminal and type

```

1 sudo mkdir /usr/local/itt/idl71/help/online_help
2 sudo ln -s /usr/local/itt/idl71/help/idl_catalog.xml \
   /usr/local/itt/idl71/help/online_help/
3 sudo ln -s /usr/local/itt/idl71/help/idlithelp.xml \
   /usr/local/itt/idl71/help/online_help/
4 sudo ln -s /usr/local/itt/idl71/help/template_help.xml \
   /usr/local/itt/idl71/help/online_help/
```

- open your Emacs and *M-x idl-rescan-asynchronously*

Happy hacking!

1.10 resources

- <http://www.idlcoyote.com/>
- <http://www.idlcoyote.com/gallery/index.html>
- <http://www.harrisgeospatial.com/docs/routines-1.html>
- http://www.harrisgeospatial.com/docs/PDF_Guides.html
- <http://idlastro.gsfc.nasa.gov/>
- <http://astropotlib.stsci.edu/>
- <http://slugidl.pbworks.com/w/page/28913791/Plotting%2520Examples>
- <http://www.astrobetter.com/wiki/tiki-index.php?page=Python+Switchers+Guide>
- url: <https://pan.baidu.com/s/1eSlZ7v0> passwd: t4x3
- url: <https://pan.baidu.com/s/1mhSk4YW> passwd: h83i

2 Call system commands

2.1 spawn

```

1  spawn, 'echo hello IDL!'
2  spawn, 'ls ~/',result
3  print, result

```

2.2 \$

```

1  $cd ~
2  $ls

```

3 Fundamentals of syntax

3.1 data types

3.1.1 get data type and size

1. help

help procedure is convenient to use in interactive mode

```

1  a = dist(256)
2  help, a

```

2. size

```

1  a = [[0, 1, 1], [2, 2, 1]]
2  print, 'Number of dimensions:', size(a, /n_dimensions)
3  print, 'Size of each dimension:', size(a, /dimensions)
4  print, 'Number of elements:', size(a, /n_elements)
5  print, 'Data type name:', size(a, /tname)
6  print, 'Data type index:', size(a, /type)

```

If no keywords are set, *SIZE* returns a vector of integer type. The first element is equal to the number of dimensions of Expression. This value is zero if Expression is scalar or undefined. The next elements contain the size of each dimension, one element per dimension (none if Expression is scalar or undefined). After the dimension sizes, the last two elements contain the type code (zero if undefined) and the number of elements in Expression, respectively.

```

1  data = dist(32)
2  result = size(data)
3  ndims = result[0]
4  if (ndims gt 0) then dims = result[1:dims] else dims = -1L
5  type = result[ndims + 1]
6  nele = result[ndims + 2]

```

The *n_elements* is commonly used within IDL programs to test whether a variable (array or otherwise) is defined.

3.1.2 type codes and names

Table 1: IDL Type Code and Names

Data type	Type code	Type name
Undefined	0	'UNDEFINED'
Byte	1	'BYTE'
Integer	2	'INT'
Long Integer	3	'LONG'
Float	4	'FLOAT'
Double precision	5	'DOUBLE'
Complex float	6	'COMPLEX'
String	7	'STRING'
Structure	8	'STRUCT'
Complex double precision	9	'DCOMPLEX'
Pointer	10	'POINTER'
Object reference	11	'OBJECT'
Unsigned integer	12	'UINT'
Unsigned long integer	13	'ULONG'
64-bit integer	14	'LONG64'
Unsigned 64-bit integer	15	'ULONG64'

3.1.3 data type and generate function

Table 2: Data types

Type	Byte number	Create	Function
byte	1	Var=oB	thisVar=Byte(variable)
int16	2	Var=o	thisVar=Fix(variable)
int32	4	Var=oL	thisVar=Long(variable)
int64	8	Var=oLL	thisVar=Long64(variable)
uint16	2	Var=oU	thisVar=UInt(variable)
uint32	4	Var=oUL	thisVar=ULong(variable)
uint64	8	Var=oULL	thisVar=Ulong64(variable)
float	4	Var=o.o	thisVar=Float(variable)
double	8	Var=o.oD	thisVar=Double(variable)
complex	8	Var=Complex(o.o,o.o)	thisVar=Complex(variable)
double complex	16	Var=Dcomplex(o.oD,o.oD)	thisVar=DComplex(variable)
string	0-32767	Var=”,Var=””	thisVar=String(variable)
pointer	4	Var=Ptr_New()	None
object	4	Var=Obj_New()	None

3.1.4 data type and array/index generation

Table 3: Array/Index generation

Type	Array	Index
byte	BytArr	BIndGen
int16	IntArr	IndGen
int32	LonArr	LIndGen
int64	Lon64Arr	L64IndGen
uint16	UIntArr	UIIndGen
uint32	ULon64Arr	ULIndGen
uint64	ULon64Arr	UL64IndGen
float	FltArr	FIndGen
double	DblArr	DIndGen
complex	ComplexArr	CIndGen
double complex	DComplexArr	DCIndGen
string	StrArr	SIndGen
pointer	PtrArr	None
object	ObjArr	None

3.1.5 vector/array slicing

```

1 vector1 = IndGen(4)
2 vector2 = vector1[0:2]
3 print, vector2

```

different from python

```

1 import numpy as np
2 vector1 = np.arange(4)

```

```
3 vector2 = vector1[0:2]
4 print("vector2 = ", vector2)
```

array slicing is just the same you can easily get a row of the array or a subarray through it's indices

3.1.6 ceil, floor, round and fix

1. ceil

The *CEIL* function returns the closest integer greater than or equal to its argument.

```
1 print, ceil(1.3)
2 print, ceil(-1.1)
3 print, ceil(3000000000.1D, /L64)
```

2. floor

The *FLOOR* function returns the closest integer less than or equal to its argument.

```
1 print, floor(1.4)
2 print, floor(-1.2)
3 print, floor(3000000000.1D, /L64)
```

3. round

The *ROUND* function rounds the argument to its closest integer.

```
1 print, round(5.5)
2 print, round(1.4)
3 print, round(-1.5)
4 print, round(3000000000.1D, /L64)
```

4. fix

The *fix* function returns the closest integer whose absolute value is less than or equal to its argument's.

```
1 print, fix(1.5)
2 print, fix(-1.5)
```

3.2 operators

3.2.1 operators and precedence

Table 4: IDL operators

Operator	Meaning	Precedence level (1 is highest)
()	parentheses	1
*	pointer dereference	2
^	exponentiation	2
*	scalar multiplication	3
#	array multiplication	3
##	matrix multiplication	3
mod	modulus	3
+	addition	4
-	subtraction and negation	4
<	minimum	4
>	maximum	4
not	boolean negation	4
eq	equal to	5
ne	not equal to	5
le	less than or equal to	5
lt	less than	5
ge	greater than or equal to	5
gt	greater than	5
and	boolean AND	6
or	boolean OR	6
xor	Boolean exclusive OR	6
:?	ternary operator	7
=	assignment	8

3.2.2 minimum and maximum operators

```

1 a = [2, 4, 6, 8]
2 b = [3, 4, 5, 6]
3 print, a < b
4 print, 3 > a
5 print, -3 < (-indgen(9))

```

3.2.3 array and matrix multiplication operators

```

1 a = indgen(3, 2) + 1
2 b = indgen(2, 3) - 1
3 print, a
4 print, b
5 c = a # b
6 print, c
7 print, transpose(c)
8 d = a ## b
9 print, d

```

3.2.4 relational and boolean operators

```

1 a = 10.0
2 b = 20.0
3 help, a gt b
4 help, a lt b

```

Table 5: True/false definitions

Data Type	True	False
Integer	Odd nonzero values	Zero and even values
Floating point	Nonzero values	Zero
String	Any string that is not null	Null string

```

1 a = [2, 4, 6, 8]
2 b = [3, 4, 5, 6]
3 print, a gt b

1 arr = byte([0, 1, 2, 3, 4])
2 print, arr
3 print, not arr
4 arr = long([0, 1, 2, 3, 4])
5 print, not arr

1 a = [0, 0, 1, 1, 2]
2 b = [0, 1, 1, 2, 2]
3 print, a or b
4 print, float(a) or float(b)
5 print, float(b) or float(a)

1 print, 4, format = '(b08)'
2 print, 5, format = '(b08)'
3 print, 4 xor 5, format = '(b08)'

```

3.2.5 ternary operator

```

1 x = 10
2 y = 5
3 z = (x lt y) ? x : y
4 print, z

```

3.3 arrays

3.3.1 array generation

1. general method

```
1 array = [[1, 2, 3], [4, 5, 6]]
```

the same as

```
1 vector = IndGen(6) + 1
2 array = Reform(vector, 3, 2)
```

2. make_array

```
1 array1 = make_array(3, 3, /integer, /index)
2 array2 = make_array(3, 3, /integer, value=8)
```

3.3.2 properties

Table 6: Functions for array properties

Function name	Return
n_elements()	number of elements
size()	array size and type information
min()	minimum array value
max()	maximum array value
mean()	mean of array values
variance()	variance of array values
stddev()	standard deviation of array values
moment()	mean, variance, skew, kurtosis, standard deviation, mean absolute deviation
total()	sum of array values

the *min* function can also return the maximum value via the *max* keyword

```

1 arr = dist(32)
2 print, min(arr), max(arr)
3 minval = min(arr, max=maxval)
4 print, minval, maxval
5 maxval = max(arr, index)
6 print, index

```

The *moment* function returns the mean, variance, skew, and kurtosis of an array

```

1 result = moment(arr)
2 print, result

```

If the optional *sdev* and *mdev* keyword are supplied, *moment* also returns the standard deviation and mean absolute deviation

```

1 result = moment(arr, sdev=sdev, mdev=mdev)
2 print, sdev, mdev

```

The *total* function returns the total of an array

```

1 arr = indgen(3, 3)
2 print, arr
3 print, total(arr)
4 print, total(arr, 1)
5 print, total(arr, 2)
6 print, total(arr, /cumulative)

```

3.3.3 location values within an array

```

1 arr = indgen(9) * 10
2 print, arr
3 index = where(arr gt 35)
4 print, arr[index]
5
6 result = arr gt 35

```

```

7 print, result
8 index = where(result)
9 print, index
10 print, arr[index]
11
12 index1 = where((arr gt 35) and ((arr mod 4) eq 0))
13 print, arr[index1]

```

You should always account for the possibility that where will not find any non zero values in the input array or expression (i.e., no values meet the prescribed conditions). In this case, where returns a scalar long equal to -1

```

1 index2 = where(arr lt -10000.0, count)
2 help, index2
3 print, count
4 index3 = where(arr gt 2.0, count)
5 print, count

```

3.3.4 reform an array

```

1 arr = indgen(6, 2)
2 print, arr
3 arr = reform(arr, 3, 4)
4 print, arr
5 col = arr[0, *]
6 col = reform(col)
7 help, col

```

To conserve the memory, the *overwrite* keyword may be used with *reform* to change the dimension information of an array without making a copy of the array

```

1 arr = dist(32)
2 help, arr
3 arr = reform(arr, 32L * 32L, /overwrite)
4 help, arr

```

3.3.5 reversing array elements

```

1 arr = indgen(5)
2 print, arr
3 print, reverse(arr)
4
5 arr = indgen(3, 3)
6 print, arr
7 print, reverse(arr)
8 print, reverse(arr, 1)
9 print, reverse(arr, 2)

```

3.3.6 rotating arrays

```

1 arr = indgen(4, 3)
2 print, arr
3 print, rotate(arr, 1)
4 print, rotate(arr, 2)
5 print, rotate(arr, 3)

```

The direction flags 0, 1, 2, 3 specify rotations of 0, 90, 180 and 270 degrees, respectively.

3.3.7 transposing arrays

```

1 arr = indgen(4, 3)
2 print, arr
3 help, transpose(arr)
4 print, transpose(arr)

5
6 arr = indgen(2, 4, 6)
7 help, transpose(arr)
8 help, transpose(arr, [1, 2, 0])

```

3.3.8 shifting arrays

```

1 arr = indgen(5)
2 print, shift(arr, 1)
3 print, shift(arr, -2)

4
5 arr = indgen(3, 3)
6 print, arr
7 print, shift(arr, 0, 1)

```

3.3.9 sorting arrays

```

1 arr = [5, 3, 9, 4, 0, 2]
2 index = sort(arr)
3 print, index
4 print, arr[index]

5
6 print, sort([1, 1, 1, 1, 1, 1])
7 print, sort([0, 0, 1, 1, 2, 2])

```

3.3.10 finding unique array values

```

1 arr = [20, 30, 20, 40, 40, 30, 50, 60, 10]
2 arr = arr[sort(arr)]
3 arr = arr[uniq(arr)]
4 print, arr

```

If you want to extract unique values from an array while preserving the order of the elements.

```

1 arr = [20, 30, 20, 40, 40, 30, 50, 60, 10]
2 sort_index = sort(arr)
3 uniq_index = uniq(arr[sort_index])
4 index = sort_index[uniq_index]
5 index = index[sort(index)]
6 print, arr[index]

```

3.3.11 array resizing

1. functions

Table 7: Functions for array resizing

Function Name	Purpose	Algorithm
rebin()	Resize an n-dimensional array by an integer multiple or factor	Default is linear interpolation for enlarging and neighborhood averaging for shrinking. Nearest-neighbor sampling is optional.
congrid()	Resize a one-, two-, or three-dimensional array to an arbitrary size (user-friendly version)	Default is nearest-neighbor sampling. Linear or cubic convolution interpolation is optional.
interpolate()	Resize a one-, two-, or three-dimensional array to an arbitrary size (generalized version).	Default is linear interpolation. Cubic convolution interpolation is optional.

2. resizing by an integer factor

```

1 arr = [20, 40, 60]
2 print, rebin(arr, 9)
3 print, rebin(arr, 9, /sample)

```

Setting the *sample* keyword (sample or sample = 1) causes nearest neighbor sampling to be used for enlarging or shrinking. If the *sample* keyword is not set (sample absent or sample = 0), then linear interpolation is used for enlarging, yielding quite different results.

When shrinking an array, *rebin* uses neighborhood averaging.

```

1 arr = indgen(9)
2 print, arr
3 print, rebin(arr, 3)

```

The *rebin* function is useful when you wish to create multidimensional grid arrays.

```

1 v = indgen(5)
2 print, v
3 print, rebin(v, 5, 3, /sample)
4 print, rebin(reform(v, 1, 5), 3, 5, /sample)
5 print, rebin(reform(v, 1, 5), 3, 5, 3, /sample)

```

3. resizing to arbitrary size

The *congrid* function enlarges or shrinking a one-, two-, or three- dimensional array to an arbitrary size. The default is to use nearest neighbor sampling.

```

1 arr = [20, 40, 60]
2 print, congrid(arr, 4)
3 print, congrid(arr, 5)
4 print, congrid(arr, 6)
5 print, congrid(arr, 7)
6 print, congrid(arr, 8)
7 print, congrid(arr, 9)
8
9 ix = 3
10 nx = 9

```

```

11 index = (float(ix) / float(nx)) * findgen(nx)
12 print, index, format='(9f6.2)'
13 index = round(index)
14 print, index, format='(9i6)'
15 print, arr[index], format='(9i6)'

```

If the optional *minus_one* keyword is set, then *congrid* uses a slightly different algorithm to compute the resampled index array. You may set *interp* keyword to use linear interpolation.

```

1 print, congrid(arr, 9, /interp)
2 print, congrid(arr, 9, /interp, /minus_one)

```

4. resizing to arbitrary size with customized interpolation

The *interpolate* function differs from *congrid* in that with *interpolate*, you specify the exact locations for which you desire interpolated values

```

1 arr = [10.0, 20.0, 30.0]
2 loc = [0.0, 0.5, 1.0, 1.5, 2.0]
3 print, interpolate(arr, loc)

```

To resize a two-dimensional array (such as an image) with *interpolate*, you must choose an algorithm to compute the locations for the interpolated array values. The *grid* keyword allows the location arrays to have a different size in each output array dimension.

```

1 a = dist(32, 32)
2 nx = 500
3 ny = 500
4 dims = size(a, /dimensions)
5 xloc = (findgen(nx) + 0.5) * (dims[0] / float(nx)) - 0.5
6 yloc = (findgen(ny) + 0.5) * (dims[1] / float(ny)) - 0.5
7 b = interpolate(a, xloc, yloc, /grid)
8 help, b
9 tvscl, b
10 c = interpolate(a, round(xloc), round(yloc), /grid)
11 tvscl, c

```

3.3.12 removing rows or columns

```

1 arr = findgen(10, 1000)
2 delrow = [16, 200, 545, 762, 998]
3 dims = size(arr, /dimensions)
4 nrows = dims[1]
5 help, nrows
6 index = replicate(1L, nrows)
7 index[delrow] = 0L
8 keeprow = where(index eq 1)
9 arr = arr[*, keeprow]
10 help, arr

```

3.4 structures

3.4.1 anonymous structures

```

1 image = {name:'Test Image', valid_range:[0.0, 100.0], $  

2     data:dist(256)}  

3 help, image  

4 help, image, /structure  

5 help, image.(0)

1 image.data = image.data * (1.0 / max(image.data))  

2 image.name = 'Normalized Test Image'  

3 image.valid_range = [0.0, 1.0]

```

3.4.2 arrays of structures

```

1 images = replicate(image, 10)  

2 help, images  

3 help, images[0]

```

3.4.3 named structures

```

1 rec = {nav_record, time:0.0, lat:0.0, lon:0.0, $  

2     heading:0.0}  

3 help, rec  

4 a = {nav_record}  

5 a.time = 12.25  

6 b = rec  

7 b.time = 13.50  

8 data = replicate({nav_record}, 100)  

9 help, data  

10 c = [a, b]  

11 help, c

```

3.4.4 working with structures

```

1 image = {name:'Test Image', valid_range:[0.0, 100.0], $  

2     data:dist(256)}  

3 print, n_tags(image)  

4 print, tag_names(image)

1 image = create_struct('data', dist(256), $  

2                     'valid_range', [0.0, 100.0])  

3 image = create_struct(image, 'date', systime())  

4 help, image.data  

5 help, image.date

```

3.5 pointers

3.5.1 functions and procedures

Table 8: Functions and procedures for pointers

Name	Purpose
ptr_new()	return a new pointer
ptrarr()	return an array of new pointers
ptr_free	free an array of new pointers
ptr_valid()	check the validity of a pointer

3.5.2 creating pointers

```

1 p = ptr_new(1.0)
2 help, p
3 print, *p
4 arr = indgen(5)
5 ptr1 = ptr_new(arr)
6 help, *ptr1
1 ptr2 = ptr_new(/allocate_heap)
2 help, *ptr2
3 *ptr2 = dist(256)
4 help, *ptr2

```

to create a pointer array

```

1 ptr = ptrarr(10, /allocate_heap)
2 *ptr[0] = dist(256)
3 *ptr[1] = indgen(10)
4 help, *ptr[0:2]

```

3.5.3 freeing pointers

```

1 ptr = ptr_new(10.0)
2 ptr_free, ptr

```

3.5.4 checking pointer validity

```

1 ptr = ptr_new(10.0)
2 print, ptr_valid(ptr)
3 ptr_free, ptr
4 print, ptr_valid(ptr)

```

3.5.5 pointer dereferencing

```

1 ptr = ptr_new(indgen(4, 4))
2 print, *ptr
3 print, (*ptr)[0, *]

```

if a pointer is included in a structure, the dereference operator is placed immediately before the structure name

```

1 rec = {flag:1L, data_ptr:ptr_new(indgen(5))}
2 print, *rec.data_ptr
3 struct_ptr = ptr_new(rec)
4 print, *(*struct_ptr).data_ptr

```

3.5.6 avoiding pointer problems

To see a list of all variables in memory that are currently referenced by a pointer, or were previously referenced by a pointer

```
1 help, /heap_variables
```

To guard against memory leakage, you can test the proposed pointer variable to see if it is already a valid pointer

```
1 if (ptr_valid(ptr) eq 1) then ptr_free, ptr
2 ptr = ptr_new(10.0)
```

To perform garbage collection, call the *heap_gc* procedure with the *ptr* and *verbose* keyword set

```
1 ptr = ptr_new(dist(256))
2 ptr = ptr_new(indgen(10))
3 ptr = ptr_new(2.0)
4 heap_gc, /ptr, /verbose
```

4 Basic I/O

4.1 standard input and output

4.1.1 procedures and functions

Table 9: procedures and functions for standard I/O

Name	Purpose
print	write formatted data to standard output
read	read formatted data from standard input
reads	read formatted data from a string
string()	write formatted data to a string

4.1.2 format

https://www.exelisvis.com/docs/format_codes.html

The syntax of an IDL format code is:

```
1 [n]FC[+][-][width]
```

where:

Table 10: format code syntax

Code	Description
n	the number of times the format code should be processed
FC	the format code
+	a + prefix before positive numbers that only valid for numeric format
-	left-justified
width	width specification

Table 11: format codes

Format Code	Description
a	character
:	terminates processing
\$	suppresses newlines in output
F,D,E,G	floating-point
B,I,O,Z	integer (B:Binary,I:Decimal,O:Octal,Z:Hexadecimal)
Q	the number of characters
quoted string and H	output string directly
T	specifies the absolute position within a record
TL	moves the position with a record to the left
TR and X	move the position within a record to the right
C()	transfers calendar data
C printf-style	an alternative syntax for specifying the format of an output
/	newline

examples:

```

1 print, format='(a)', 'hello world'
2 print, format='(a7)', 'hello world'
3 print, format='(i12)', 300
4 print, format='(i12.8)', 300
5 print, format='(i08)', 300
6 print, format='(b)', 7
7 print, format='(o)', 7
8 print, format='(z)', 12
9 print, format='(3f12.3)', findgen(3) * 500
10 print, format='(f12.3, f10.2, f10.1)', findgen(3) * 500
11 print, format='(3f12.3)', findgen(9) * 500
12 print, format='(3(f12.3, :, ", "))', findgen(9) * 500
13 print, format='(3(e12.3, :, ", "))', findgen(9) * 500
14 print, format='(3(e+12.3, :, "$"))', findgen(9) * 500 - 1000
15 print, format='(g12.4)', 10
16 print, format='(g12.4)', 10000000

1 .run
2 pro format_test
3   name = ''
4   age = 0
5   print, format='($, "Enter your name")'
6   read, name
7   print, format='("Enter your age")'
8   read, age
9   print, format='("You are ", i0, " years old, ", a0)', age, name
10 end
11
12 format_test

1 read, format='(q)', charnumber
2 print, charnumber

1 print, format='("Value: ", I0)', 23
2 print, format='(7HValue: , I0)', 23

```

```

1 print, format='("First", 20X, "Last", T10, "Middle")'
2 print, format='("First", 20X, "Last", TL15, "Middle")'
3 print, format='("First", TR15, "Last")'

1 print, systime()
2 print, systime(/utc)
3 print, systime(1)
4 print, systime(/julian)
5 curtime = systime(/julian)
6 print, format='(C0)', curtime
7 print, format='(CC(YI, "-", CMOI, "-", CDI))', curtime
8 print, format='(C(CHI, ":"), CMI, ":"), CSI)', curtime

```

4.2 working with files

4.2.1 procedures and functions

Table 12: procedures and functions for working with files

Name	Purpose
openr	open an existing file for read only
openw	open a new file for read and write, overwrite if existed
openu	open an existing file for updating,i.e.,read and write
findfile()	return names of files in the current directory
dialog_pickfile()	graphical file selector
fstat()	return information about an open file
eof()	check for end-of-file
close	close a file
free_lun	free a logical unit number and close a file

4.2.2 opening files

Read and write operations are performed using the logical unit number to reference the file.

a specific logical unit number, which may be set to a scalar long value in the range of 1 to 99

```

1 lun = 201
2 openw, lun, 'test.dat'
3 free_lun, lun

1 openr, 5, filename
2 close, 5

```

assign a free logical unit, an unused logical unit number in the range of 100 to 128 will be assigned

```

1 openw, lun, 'test.dat', /get_lun
2 free_lun, lun

```

4.2.3 selecting a file

```

1 list = findfile()
2 print, list

1 list = findfile('*.*dat')
2 list = findfile('*.*dat', count=numfiles)

1 file = dialog_pickfile()
2 file = dialog_pickfile(filter='*.dat')
3 file = dialog_pickfile(filter='*.pro', /multiple)
4 directory = dialog_pickfile(/directory)

```

most of the time, if we include a *Dialog_Pickfile* in a program, we just check for the null string returned from the dialog and return out of the program

```

1 file = dialog_pickfile(file='*.jpg', path=!dir)
2 if file eq '' then return

```

4.2.4 filepath

```

1 galaxy = filepath('galaxy.dat', subdirectory=['examples', 'data'])

1 cd, current=thisdir
2 galaxy = filepath('galaxy.dat', root_dir=thisdir, $
3                   subdirectory='coyote')

```

4.2.5 obtaining information about files

1. specific lun

```

1 file = filepath('hurric.dat', subdir='examples/data')
2 openr, lun, file, /get_lun
3 info = fstat(lun)
4 help, info.size, info.read, info.name
5 free_lun, lun

```

2. all luns

```
1 help, /files
```

if you want to close all files

```
1 close, /all
```

4.2.6 reading and writing formatted (ASCII) files

1. reading a formatted file

if the size data is known

```

1 openr, lun, 'input.dat', /get_lun
2 data = lonarr(10,100)
3 readf, lun, data, format='(10i6)'
4 free_lun, lun

```

if you want to read a formatted file that has a known format but an unknown length,e.g.,in the following format:

```
22:02:13.34UTC -29.93455 -116.0234W 234.2
22:02:14.33UTC -29.9357S -116.0232W 234.1
22:02:15.34UTC -29.9362S -116.0230W 234.1
22:02:16.35UTC -29.9367S -116.0228W 234.2
22:02:17.36UTC -29.9373S -116.0226W 234.2
22:02:18.34UTC -29.9378S -116.0224W 234.3
```

```

1  FUNCTION READ_POSITION, FILE, MAXREC=MAXREC
2
3      ;; check argument
4      if (n_elements(file) eq 0) then $
5          message, 'Argument FILE is undefined'
6      if (n_elements(maxrec) eq 0) then $
7          maxrec = 10000L
8
9      ;; open input file
10     openr, lun, file, /get_lun
11
12     ;; define record structure and create array
13     fmt = '(2(i2, 1x), f5.2, 4x, f8.4, 2x, f9.4, 2x, f5.1)'
14     record = {hour:0L, min:0L, sec:0.0, $
15                lat:0.0, lon:0.0, head:0.0}
16     data = replicate(record, maxrec)
17
18     ;; read record until end-of-file reached
19     nrecords = 0L
20     recnum = 1L
21     while (eof(lun) ne 1) do begin
22
23         ;; read this record (jumps to bad_rec: on error)
24         on_ioerror, bad_rec
25         error = 1
26         readf, lun, record, format=fmt
27         error = 0
28
29         ;; store data for this record
30         data[nrecords] = record
31         nrecords = nrecords + 1
32
33         ;; check if maximum record count exceeded
34         if (nrecords eq maxrec) then begin
35             free_lun, lun
36             message, 'Maximum record reached: increase MAXREC'
37         endif
38
39         ;; check for bad input record
40         bad_rec:
41         if (error eq 1) then $
42             print, 'Bad data at record', recnum
43             recnum = recnum + 1
44
45     endwhile
46
47     ;; close input file
```

```

48   free_lun, lun
49
50   ;; trim data array and return it to caller
51   data = data[0 : records - 1]
52   return, data
53
54 END

1 result = read_position('position.dat')
2 print, result[0].sec, result[0].lat

```

The *on_ioerror* procedure is used to handle any errors when reading from the input file, such as a record that with an invalid format.

2. writing a formatted file

The *printf* procedure writes data to a formatted file that has been opened for writing or updating.

```

1 data = lindgen(10, 100)
2 openw, lun, 'output.data', /get_lun
3 printf, lun, data, format='(10i6)'
4 free_lun, lun

```

If you use *printf* (instead of *print*) in your program, it's easy to send the printed output to the screen by default, or to an output file if required. This is possible because the logical unit number **-1** is reserved for standard output.

```

1 PRO REDIRECT, OUTFILE=OUTFILE
2
3   ;; Select logical unit for output
4   if (n_elements(outfile) eq 1) then begin
5     openw, outlun, outfile, /get_lun
6   endif else begin
7     outlun = -1
8   endelse
9
10  ;; printf is used in the program for output
11  printf, outlun, 'Starting program execution at', system()
12
13  ;; Body of the program goes here
14
15  ;; Close the output file if needed
16  if (outlun gt 0) then free_lun, outlun
17
18 END

```

Formatted output to a file is limited to 80 columns by default. To increase the default output width, use the *width* keyword when the *openw* or *openu* procedure is called

```

1 openw, lun, 'wide.dat', /get_lun, width=150
2 printf, lun, findgen(100), format='(10f10.2)'
3 free_lun, lun

```

4.2.7 reading and writing unformatted (binary) files

1. procedures and functions

Table 13: procedures and functions for unformatted I/O

Name	Purpose
readu	read data from an unformatted file
point_lun	reposition the file pointer in an open file
writeu	write data to an unformatted file
assoc()	associate an logical unit number with a variable
save	save variable in a portable IDL-specific format
restore	restore variables from a save file

2. reading an unformatted file (single data type)

If an unformatted file contains only one kind of binary data (e.g., byte), and the file size is known in advance, then reading the contents of the file is straightforward.

```

1 file = filepath('ctscan.dat', subdir='examples/data')
2 openr, lun, file, /get_lun
3 data = bytarr(256, 256)
4 readu, lun, data
5 free_lun, lun
6 tvscl, data

```

If the size of the file is not known in advance, but the size of each record is known

```

1 file = filepath('hurric.dat', subdir='examples/data')
2 openr, lun, file, /get_lun
3 info = fstat(lun)
4 ncols = 440L
5 nrows = info.size / ncols
6 data = bytarr(ncols, nrows)
7 readu, lun, data
8 free_lun, lun
9 tvscl, data

```

3. reading an unformatted file (mixed data type)

Unformatted data files may contain more than one data type per record.

```

1 FUNCTION READ_WEATHER, FILE
2
3     ;; Check argument
4     if (n_elements(file) eq 0) then $
5         message, 'Argument FILE is undefined'
6
7     ;; Open the input file
8     openr, lun, file, /get_lun
9
10    ;; Define record structure, and create data array
11    record = {fid:0L, year:0, month:0, day:0, hour:0, minute:0, $
12              pres:0.0, temp:0.0, dewp:0.0, speed:0.0, dir:0.0}
13    data = replicate(record, 24L * 60L)
14

```

```

15    ;; Read records from the file until EOF
16    nrecords = 0
17    while (eof(lun) ne 1) do begin
18        readu, lun, record
19        data[nrecords] = record
20        nrecords = nrecords + 1
21    endwhile
22
23    ;; Close the input file
24    free_lun, lun
25
26    ;; Trim the data array, and return it to caller
27    data = data[0 : nrecords - 1]
28    return, data
29
30 END

```

4. reading a Fortran-77 unformatted file

Files that are written in Fortran-77 unformatted sequential access mode have extra information attached to the beginning and end of each data record. For example, consider the following Fortran-77 program:

```

1      program unformatted_test
2 !     Fortran-77 program to produce unformatted output
3     implicit none
4     integer :: a(10)
5     real :: b(5)
6     integer :: index
7     do index = 1, 10
8         a(index) = index
9     end do
10    do index = 1, 5
11        b(index) = real(index * 10)
12    end do
13    open(30, file='unformatted_test.dat', form='unformatted')
14    write(30) a
15    write(30) b
16    close(30)
17 end

```

to read the contents of the output file

```

1  openr, lun, 'unformatted_test.dat', /get_lun, /f77_unformatted
2  a = lonarr(10)
3  b = fltarr(5)
4  readu, lun, a
5  readu, lun, b
6  free_lun, lun
7  print, a, format='(10i4)'
8  print, b, format='(5f5.1)'

```

5. repositioning the file pointer

The file pointer points to the position in the file where the next read or write request will begin. Every time a read or write request is issued, the file pointer is moved the appropriate number of bytes further along in the file.

```

1 openr, lun, 'tapefile.dat', /get_lun
2 point_lun, lun, 1024L
3 data = bytarr(512, 512)
4 readu, lun, data
5 free_lun, lun

```

to get the current position of the file pointer

```

1 info = fstat(lun)
2 position = info.cur_ptr

```

6. writing binary data to an unformatted file

When writing binary data to a file, it can be helpful to first write a header record that records the size and type of the data contained in the file.

```

1 data = dist(256)
2 openw, lun, 'dist.dat', /get_lun
3 info = size(data)
4 info_size = n_elements(info)
5 header = [info_size, info]
6 writeu, lun, header
7 writeu, lun, data
8 free_lun, lun

```

to read the file, you would first read the header, and then use the *make_array* function to create an array of the correct size and type to hold the data:

```

1 openr, lun, 'dist.dat', /get_lun
2 info_size = 0L
3 readu, lun, info_size
4 info = lonarr(info_size)
5 readu, lun, info
6 data = make_array(size=info)
7 readu, lun, data
8 free_lun, lun

```

7. programs to write and read portable binary data

The *binread* procedure allows you to read one or more arrays from a binary unformatted file created by *binwrite* in first-in/first-out order. The "magic" value allows *binread* to determine whether or not byte swapping is required

```

1 PRO BINWRITE, LUN, DATA
2
3   ;; Check arguments
4   if (n_elements(lun) eq 0) then $
5     message, 'LUN is undefined'
6   if (n_elements(data) eq 0) then $
7     message, 'DATA is undefined'
8
9   ;; Check that file is open
10  result = fstat(lun)
11  if (result.open eq 0) then $
12    message, 'LUN does not point to an open file'
13
14  ;; Check that data type is allowed

```

```

15      type_name = size(data, /tname)
16      if (type_name eq 'STRING') or $
17          (type_name eq 'STRUCT') or $
18          (type_name eq 'POINTER') or $
19          (type_name eq 'OBJREF') then $
20              message, 'DATA type is not supported'
21
22      ;; Check that DADA is an array
23      if (size(data, /n_dimensions) lt 1) then $
24          message, 'DATA must be an array'
25
26      ;; Create header array
27      magic_value = 123456789L
28      info = size(data)
29      info_size = n_elements(info)
30      header = [magic_value, info_size, info]
31
32      ;; Write header and data to file
33      writeu, lun, header
34      writeu, lun, data
35
36 END

1 PRO BINREAD, LUN, DATA
2
3     ;; Check arguments
4     if (n_elements(lun) eq 0) then $
5         message, 'LUN is undefined'
6     if (arg_present(data) eq 0) then $
7         message, 'DATA cannot be modified'
8
9     ;; Check that file is open
10    result = fstat(lun)
11    if (result.open eq 0) then $
12        message, 'LUN does not point to an open file'
13
14    ;; Read magic value
15    magic_value = 0L
16    readu, lun, magic_value
17
18    ;; Devode magic value to see if byte swapping is required
19    swap_flag = 0
20    if (magic_value ne 123456789L) then begin
21        magic_value = swap_endian(magic_value)
22        if (magic_value eq 123456789L) then begin
23            swap_flag = 1
24        endif else begin
25            message, 'File was not written with BINWRITE'
26        endelse
27    endif
28
29    ;; Read the header, swapping if necessary
30    info_size = 0L
31    readu, lun, info_size
32    if swap_flag then info_size = swap_endian(info_size)
33    info = lonarr(info_size)
34    readu, lun, info
35    if swap_flag then info = swap_endian(info)

```

```

36
37      ;; Read the data, swapping if necessary
38      data = make_array(size=info)
39      readu, lun, data
40      if swap_flag then data = swap_endian(temporary(data))
41
42 END

```

example of usage

```

1  a = dist(256)
2  b = lindgen(1000)
3  openw, lun, 'test.dat', /get_lun
4  binwrite, lun, a
5  binwrite, lun, b
6  free_lun, lun
7  openr, lun, 'test.dat'
8  binread, lun, a
9  binread, lun, b
10 free_lun, lun
11 help, a, b

```

8. reading binary data via an associated variable

```

1  file = filepath('people.dat', subdir='examples/data')
2  openr, lun, file, /get_lun
3  chunk = bytarr(192, 192)
4  image = assoc(lun, chunk)
5  tvscl, image[0]
6  tvscl, image[1]
7  free_lun, lun

```

5 Basic programming

5.1 defining and compiling programs

5.1.1 procedures

- A main procedure

```

1  print, 'Hello world'
2  END

```

- A procedure with no arguments:

```

1  PRO HELLO
2    print, 'hello world'
3  END

```

- A procedure with three parameters

```

1  PRO READ_IMAGE, IMAGE, DATA, TIME
2    ;; statements...
3  END

```

- A procedure with three parameters and two optional keywords

```

1 PRO PRINT_IMAGE, IMAGE, DATA, TIME, LANDSCAPE=LANDSCAPE, $
2           COLOR=COLOR
3   ;; statements
4 END

```

5.1.2 functions

- A function with one parameter

```

1 FUNCTION F_TO_C, DEG_F
2   return, (deg_f - 32.0) * (5.0 / 9.0)
3 END

```

- A function with three parameters and two optional keywords

```

1 FUNCTION GET_BOUNDS, IMAGE, LAT, LON, POLAR= POLAR, $
2           NORMAL=NORMAL
3   ;; statements...
4 END

```

5.1.3 naming source files

The standard format for naming an IDL source file is the procedure or function name followed by a . pro extension (e.g., imdisp.pro).

5.1.4 manual compilation

```

1 .compile hello.pro
2 .compile hello.pro imdisp.pro sds_read.pro

```

It is also possible to manually compile procedures or functions with the executive commands *.run* or *.rnew*. If the source file named in the *.run* or *.rnew* command is a "main program", the "main program" will be executed.

5.1.5 automatic compilation

IDL automatically compiles procedures and functions as they are needed at runtime.

5.1.6 returning to the main level after an error

If an error in an IDL program causes execution to stop, by default, IDL halts inside the procedure that causes the error.

for example

```

1 FUNCTION SUM, ARG1, ARG2
2   return, arg1 + arg2
3 END
4
5 PRO HELLO, A , B
6   print, 'Hello world'
7   print, 'The sum of A and B is ', sum(a, b)
8 END

```

To compile and run the program

```

1 .compile hello
2 a = 3
3 b = 15
4 hello, a, b

```

However, if the program is called when an argument is undefined

```

1 hello, a, bbb

```

The error message shows that IDL halted inside the *product* function at line 2 of the source file (hello.pro).

Because execution halted inside the *product* function, only the variables defined in this function are now available at the command line.

```

1 help, arg1
2 help, a

```

In addition, you will see warning messages if you try recompiling the source file after an execution halt.

After examining any variables in the failed function or procedure, use the *retall* command to return to the main level.

```

1 retall
2 help, a
3 help, arg1

```

5.2 control statements

5.2.1 IDL control statements

Table 14: IDL control statements

Statement	Purpose
if	If a condition is true, execute statement(s)
case	Select one case to execute from a list of cases
for	For a specified number of times, execute a statement loop
while	While a condition is true, execute a statement loop
repeat	Repeat a statement loop until a condition is true
return	Return control to the calling function or procedure
goto	Go to a label
switch	Branch to a case in a list of cases
break	Break out of a loop, <i>case</i> statement, or <i>switch</i> statement
continue	Continue execution on the next iteration of a loop

5.2.2 if statement

```

1 if condition then statement
2
1 if condition then begin
2   statement(s)
3 endif

```

```

1 if condition then statement else statement
1 if condition then begin
2   statement(s)
3 endif else begin
4   statement(s)
5 endelse

```

In all forms , *condition* is a scalar expression that evaluates to true or false.

5.2.3 case statement

```

1 case expression of
2   exp1:
3     statement
4   exp2: begin
5     statement(s)
6   end
7   else: statement
8 endcase

```

Here, *expression*, *exp1*, *exp2* and *exp3* are scalar expression.

The *case* statement can also be used in conjunction with relational and Boolean operators.

```

1 case 1 of
2   cond1: statement
3   cond2: begin
4     statement(s)
5   end
6   else: statement
7 endcase

```

Here *cond1* and *cond2* are expressions containing relational operators that evaluate to either true or false, such as (x gt 100).

5.2.4 for statement

```

1 for i = v1, v2 do statement
1 for i = v1, v2, inc do statement
1 for i = v1, v2, inc do begin
2   statements(s)
3 endfor

```

If the default *int* type is used for the loop start variable, large loop end values can cause problems

```
1 for i = 0, 40000, 1000 do print, i
```

The loop end value of 40,000 is automatically interpreted as a *long* since it is greater than 32,767, and therefore a mismatch exists between the types of the loop start and end values. For this reason, *long* loop control variables are recommended.

```
1 for i = 0L, 40000L, 1000L do print, i
```

Floating-point loop variables are permitted, but are not recommended. If you need to create floating-point variables inside a loop, the preferred approach is to create a float variable from the loop control variable.

```
1 for i = 1L, 5L, 1L do print, 0.1 * float(i)
```

5.2.5 ***while*** statement

```

1 while condition do statement
2
3   while condition do begin
4     statement(s)
5   endwhile

```

5.2.6 ***repeat*** statement

```

1 repeat statemnt until condition
2
3   repeat begin
4     statement(s)
5   endrep until condition

```

5.2.7 ***return*** statement

In general, you should **try** to limit each function to one return statement.

5.2.8 ***goto*** statement

Jump to a specific location

```
1 goto, label
```

The destination is specified by *label*

```
1 label:
```

5.2.9 ***switch*** statement

```

1 switch expression of
2   exp1:
3   exp2: statement
4   exp3: begin
5     statement(s)
6   end
7   else: statement
8 endswitch

```

Here, *expression*, *exp1*, *exp2* and *exp3* are scalar expressions.

5.2.10 ***break*** statement

The break statement causes an immediate exit from a *for*, *while* , or *repeat* loop, or a case or switch statement. Control is transferred to the next statement after the end of the loop, or an immediate exit from a case or switch statement.

5.2.11 ***continue*** statement

The continue statement causes the next iteration of a for, while, or repeat loop to be executed. Any statements remaining in the current loop iteration are skipped.

5.3 arguments and keywords

5.3.1 argument passing

Table 15: Arguments passed by reference versus value

Passed by reference	Passed by value
Scalars	Constants
Arrays	Indexed subarrays
Structures	Structure elements
Undefined variables	System variables
	Expressions

5.3.2 extra keywords

When keywords are passed to a procedure or function, it is sometimes necessary to pass these keywords to another routine from within the called routine.

```

1 PRO E PLOT, X, Y, NAME=NAME, _EXTRA=EXTRA_KEYWORDS
2
3   ;; Check arguments
4   if (n_params() ne 2) then $
5     message, 'Usage: E PLOT, X, Y'
6   if (n_elements(x) eq 0) then $
7     message, 'Argument X is undefined'
8   if (n_elements(y) eq 0) then $
9     message, 'Argument Y is undefined'
10  if (n_elements(name) eq 0) then name = 'Faxiang Zheng'
11
12  ;; Plot the data
13  plot, x, y, _extra=extra_keywords
14
15  ;; Print name and date on plot
16  date = systime()
17  xyouts, 0.0, 0.0, name, align=0.0, /normal
18  xyouts, 1.0, 0.0, date, align=1.0, /normal
19
20 END

```

then in your IDL shell,

```

1 x = findgen(200) * 0.1
2 eplot, x, sin(x), xrange=[0, 15], $
3           xtitle='X', ytitle='SIN(X)'

```

5.3.3 extra keywords precedence

A keyword passed to a procedure or function via the extra keyword method takes precedence when a keyword with the same name is passed in the argument list.

```

1 PRO K PLOT, _EXTRA=EXTRA_KEYWORDS
2
3   plot, indgen(10), psym=2, _extra=extra_keywords
4
5 END

```

When this procedure is called with no keywords, the *plot* procedure uses plot symbol 2. If the *psym* keyword is passed when *kplot* is called, the *plot* procedure uses the value of *psym* passed via the extra keyword structure and ignore the explicit setting of *psym=2* in the argument list.

```
1 kplot
2 kplot, psym=4
```

If you wish to use a keyword inside a procedure or function that would otherwise be passed via the extra keyword method, you must include the keyword in the procedure or function definition.

```
1 PRO KPLOT, PSYM=PSYM, _EXTRA=EXTRA_KEYWORDS
2
3   if (n_elements(psym) eq 0) then psym = 2
4   print, 'PSYM = ', psym
5   plot, indgen(10), psym=psym, _extra=extra_keywords
6
7 END
```

5.3.4 extra keyword passing

If the *_ref_extra* keyword is used instead of *_extra* in the procedure or function definition, then the keywords are passed by reference and therefore can be modified by the routine to which they are passed.

```
1 FUNCTION SQUARE, ARG, POSITIVE=POSITIVE
2   result = arg ^ 2
3   if (arg gt 0) then positive = 1 else positive = 0
4   return, result
5 END
6
7 PRO TEST_VAL, INPUT, OUTPUT, _EXTRA=EXTRA_KEYWORDS
8   output = square(input, _extra=extra_keywords)
9 END
10
11 PRO TEST_REF, INPUT, OUTPUT, _REF_EXTRA=EXTRA_KEYWORDS
12   output = square(input, _extra=extra_keywords)
13 END
14
15 .compile test_extra
16 test_val, 2.0, output, positive=pos_flag
17 help, pos_flag
18
19 test_ref, 2.0, output, positive=pos_flag
20 help, pos_flag
```

5.4 checking parameters and keywords

5.4.1 routines

Table 16: Routines for checking arguments

Name	Purpose
n_params()	Returns number of parameters passed (not including keywords)
n_elements()	Returns number of elements in a variable (zero means variable is undefined)
size()	Returns size and type information about a variable
arg_present()	Returns true if an argument was present and was passed by reference
message	Print a message and halts execution

5.4.2 checking input parameters and keywords

Consider a hypothetical plotting procedure named *myplot.pro*, which is defined as follows

1 MYPLOT, X, Y, POSITION=POSITION

The parameters *x* and *y* are one-dimensional vectors. Both parameters are required, and they must be the same size. The *position* keyword must be one-dimensional vector with four elements if supplied; otherwise the default value is [0.0, 0.0, 1.0, 1.0]. The following checks are appropriate for this routine.

- Was the correct number of input parameters passed?

```
1 if (n_params() ne 2) then $
2   message, 'Usage: MYPLOT, X, Y'
```

- Is each input parameter defined?

```
1 if (n_elements(x) eq 0) then $
2   message, 'X is undefined'
3 if (n_elements(y) eq 0) then $
4   message, 'Y is undefined'
```

- Does each input parameter have the correct number of dimensions?

```
1 if (size(x, /n_dimensions) ne 1) then $
2   message, 'X must be a 1D array'
3 if (size(y, /n_dimensions) ne 1) then $
4   message, 'Y must be a 1D array'
```

- Are the input parameters the same size?

```
1 if (n_elements(x) ne n_elements(y)) then $
2   message, 'X and Y must be same size'
```

- If a keyword was not passed, is a default value set?

```

1 if (n_elements(position) eq 0) then $
2   position = [0.0, 0.0, 1.0, 1.0]

```

6. If a keyword was passed, does it have the correct number of elements?

```

1 if (n_elements(position) ne 4) then $
2   message, 'POSITION must have 4 elements'

```

5.4.3 don't modify input parameters

You should avoid modifying input parameters.

5.4.4 checking boolean keywords

Boolean keywords are checked using the *keyword_set* function, which returns true if the argument is defined, **and** is a nonzero scalar, **or** an array of any size, type, or value. For example, the *myplot* procedure shown previously could an *erase* keyword which causes the screen to be erased if it is set to true.

```
1 if keyword_set(erase) then erase
```

or

```

1 erase = keyword_set(erase)
2 if (erase) then erase

```

In this case, if *erase* is a nonzero scalar, or an array of any size, type, or value, it will be converted to a scalar int equal to 1, which will be returned to the caller.

Another case where you may wish to use a Boolean keyword within a procedure or function is to pass it to another routine that accepts a Boolean keyword with the same name.

```
1 plot, x, y, erase=keyword_set(erase)
```

5.4.5 checking output parameters and keywords

Output arguments must be passed by reference if they are to be created or modified within a called routine and returned to the caller. The *arg_present* function is used to check for arguments that can be modified inside a called routine. If a parameter or keyword was present in the calling sequence, and the parameter or keyword was passed by reference, *arg_present* returns true.

```

1 if arg_present(xrange) then $
2   xrange = [min(x) - 1.0, max(x) + 1.0]

```

In the case of mandatory output parameters, *arg_present* allows you to detect parameters that cannot be modified.

```

1 if (arg_present(slope) ne 1) then $
2   message, 'SLOPE cannot be modified'

```

5.5 scripts

The simplest IDL program is writing multiple command lines to a file, just the same as the commands used in the interactive IDL shell. For example, put the following commands into *hello.pro*

```

1 print, 'hello world'
2 print, 'I love IDL'
3 for index = 0, 10 do print, index

```

Then, you can call the command in the file by

```
1 @hello
```

in your IDL shell.

However, the following multiple lines for loop is not permitted in a script because it contains a statement block

```

1 for index = 0, 10 do begin
2   print, index
3 endfor

```

It is possible to bypass this behavior by including multiple statements on one line separated by the & character

```
1 for index = 0, 10 do begin & print, index & endfor
```

This method of defining a loop should only be used in a script or on the command line.

5.6 include files

An include file contains a sequence of IDL statements that are inserted in a procedure or function at compile time. Include files may contain any statement that is legal in a procedure or function, including multiline statement blocks. For example, the following physical constants could be contained in an include file named *fundamental_constants.inc*

```

1 planck_constant = 6.6260755d-34      ; Joule second
2 light_speed = 2.9979246d+8          ; meters per second
3 boltzmann_constant = 1.380658d-23    ; Joules per Kelvin
4 rad_c1 = 2.0d0 * planck_constant * light_speed^2
5 rad_c2 = (planck_constant * light_speed) / boltzmann_constant

```

To include the file in a function named *planck.pro* that computes monochromatic Planck radiance given wavenumber and temperature

```

1 FUNCTION PLANCK, V, T
2   @fundamental_constants.inc
3   vs = 10d2 * v
4   return, vs^3 * ((rad_c1 * 10D5) / $
5                   (exp(rad_c2 * (vs / t)) - 1.0D0))
6 END

```

The include file is silently inserted into the function at compile time

5.7 journal

A journal of an IDL session can be created by calling the *journal* procedure

```
1 journal
```

The default journal file *idlsave.pro* is created in the current directory. The *journal* procedure accepts one optional parameter specified the path and name of the journal file.

```
1 journal, 'mydemo.pro'
```

Journal ends when *journal* is called without any arguments, or when the IDL session ends. To replay a journal file, simply use the 'at' character

```
1 @mydemo
```

5.8 global variables

5.8.1 read-only system variables

Table 17: Commonly used read-only system variables

System variable	Meaning
<code>!dtor</code>	Degrees to radians conversion factor
<code>!radeg</code>	Radians to degrees convolution factor
<code>!pi</code>	π (single precision)
<code>!dpi</code>	π (double precision)
<code>!version.release</code>	IDL version number (string)
<code>!version.os_family</code>	Platform identifier
<code>!d.flags</code>	Bitmask containing information about the current graphics device
<code>!d.n_colors</code>	Number of possible colors in palette
<code>!d.name</code>	Name of the current graphics device
<code>!d.table_size</code>	Size of current graphics window
<code>!d.window</code>	Index of current graphics window
<code>!d.x_vsize</code>	Viewable horizontal size of current graphics device (in pixels)
<code>!d.y_vsize</code>	Viewable vertical size of current graphics device (in pixels)
<code>!x.window</code>	Horizontal coordinates of current plot (normalized units)
<code>!y.window</code>	Vertical coordinates of current plot (normalized units)

5.8.2 writable system variables

Table 18: Commonly used writable system variables

System variables	Meaning
<code>!p.multi</code>	Configures multipanel plots
<code>!p.font</code>	Selects default font for plotted characters
<code>!order</code>	Controls image display order (bottom up or top down)
<code>!path</code>	The current search path for procedures and functions
<code>!except</code>	Controls math error behavior

5.8.3 user-defined system variables

```

1 defsv, 'plot_info', 10.0
2 help, !plot_info

```

By default, system variables created by *defsv* are writable, but their type and size cannot be changed

```

1 !plot_info = -32.0
2 help, !plot_info

```

However, read-only system variables can be created by adding an optional nonzero argument to *defsv*, as shown in the following example, where the speed of light is defined in meters per second

```

1 defsv, '!speed_of_light', 2.9979246d+8, 1
2 help, !speed_of_light
3 !speed_of_light = 123.45

```

To remove user-defined system variables, you can either exit IDL and start a new session, or issue the *.reset_session* executive command at the IDL command line to reset the session.

```

1 .reset_session
2 help, !plot_info

```

To check for the existence of a system variable, the *exist* keyword is used in conjunction with *defsv* to return a logical variable

```

1 defsv, '!plot_info', 20.0
2 defsv, '!plot_info', exist=exist
3 help, exist
4 defsv, '!plot_nothing', exist=exist
5 help, exist

```

5.9 error handling

5.9.1 on_error

When IDL encounters a fatal error in a procedure or function, the default behavior is to stop execution in the routine that cause the error. This default behavior may be modified by calling the *on_error* procedure, which accepts a single scalar argument specify the default error behavior.

Table 19: Arguments values for *on_error*

Value	Behavior when error occurs
0	Stop the routine that cause the error (default)
1	Return to the main level
2	Return to the caller of the routine that cause the error
3	Return to the routine that cause the error

5.9.2 intercepting errors

If you wish to manually intercept error and provide your own error-handling code, the *catch* procedure can be used.

The call to `ca tch` establishes the error handler and creates the variable `error_status`, which is set to zero. In the event of a subsequent error, `error_status` is set to a nonzero error code.

```

1  FUNCTION NCDF_ISNCDF, FILE
2
3      ;; Check arguments
4      if (n_params() ne 1) then $
5          message, 'Usage: RESULT = NCDF_ISNCDF(FILE)'
6      if (n_elements(file) eq 0) then $
7          message, 'Argument FILE is undefined'
8
9      ;; Establish error handler
10     catch, error_status
11     if (error_status ne 0) then return, 0
12
13     ;; Attempt to open the file
14     ncid = ncdf_open(file)
15
16     ;; Close the file and return to caller
17     ncdf_close, ncid
18     return, 1
19
20 END

```

If you wish to obtain further information about an error that was handled by *catch*

```

1  if (error_status ne 0) then begin
2      help, /last_message, output=errtext
3      print, errtext[0]
4      return, 0
5  endif

```

5.9.3 math errors

IDL is able to detect math errors such as overflow, underflow, and divide-by-zero. In these cases, IDL substitutes the special floating-point values *NaN* (not a number) or *Inf* (infinity) for the value that caused the error

```

1  print, sqrt(-1.0)
2  print, 1.0/0.0

```

The default behavior is to continue execution and print a warning message when control is returned to the command line. To change the default behavior, you can change the system variable `!except`,

Table 20: Values for system variable `!except`

Value	Behavior when math error occurs
0	Do not print a warning
1	Print a warning when control returns to the command line (default)
2	Print a warning when the exception occurs

If you wish to check the math error status without printing a warning message, set `!except` to zero and call the `check_math` function

The `finite` function may be used to check for the presence of `NaN` and `Inf`.

```
1 arr = [0.0, 1.0, sqrt(-1.0), 4.0, 1.0 / 0.0]
2 print, finite(arr)
```

To locate non-finite values and set them to a missing value, use the `finite` function in conjunction with `where` to return the indices of the non-finite numbers

```
1 missing = -999.9
2 index = where(finite(arr) eq 0, count)
3 if (count gt 0) then arr[index] = missing
4 print, arr
```

5.10 efficient programming

5.10.1 conserving memory

The `temporary` function returns a temporary copy of a variable, and sets the original variable to “undefined”. This function can be used to conserve memory when performing operations on large arrays, as it avoids making a new copy of results that are only temporary. In general, the `temporary` routine can be used to advantage whenever a variable containing an array on the left hand side of an assignment statement is also referenced on the right hand side.

```
1 ;; generate an array
2 a = dist(500, 500)
3 ;; normal way, requires more space
4 a = a + 1.0
5 ;; better way, requires no additional space
6 a = temporary(a) + 1.0
```

If multiple operations are performed on the array, they can be performed one step at a time to conserve memory

```
1 ;; normal way
2 a = a * (1.0 / max(a))
3 ;; better way
4 maxval = max(a)
5 a = temporary(a) / maxval
```

When an array is replaced by a sub-array

```
1 ;; generate an array
2 a = dist(256)
3 ;; normal way
4 a = a[0:63, 0:63]
5 ;; better way
6 a = (temporary(a))[0:63, 0:63]
```

5.10.2 using efficient methods

If you want to find the sum of all elements greater than 100.0 in a frequency distribution array with 500 columns and 1,000 rows.

```

1  ;; generate an array
2  a = dist(500, 1000)
3  ;; normal way
4  sum = 0.0
5  for i = 0L, n_elements(a) - 1L do $
6    if (a[i] gt 100.0) then sum = sum + a[i]
7  ;; a better way
8  index = where(a gt 100.0, count)
9  if (count gt 0) then sum = total(a[index])
10 ;; another better way
11 sum = total(a * (a gt 100.0))

```

To time an operation

```

1  ;; measure time
2  t0 = systime(1) & sum = total(a * (a gt 100.0)) & $
3  t1 = systime(1)
4  print, t1 - t0

```

6 Window

6.1 create window

```

1  window
2  window, 10
3  window, /free

```

6.2 current window number

```

1  print, !D.Window

```

6.3 set specific window to current

```

1  wset, 10

```

6.4 delete windows

6.4.1 delete a window

```

1  wdelete, 10

```

6.4.2 delete all windows

```

1  WHILE !D.Window NE -1 DO Wdelete, !DWindow

```

6.5 window size and position

```

1  window, 1, xsize=200, ysize=300, xpos=75, ypos=150

```

6.6 show window

```
1 wshow, 1
```

6.7 window title

```
1 window, title='fuck'
```

6.8 erase current window

```
1 erase
```

6.9 device independent graphics windows

```
1 cgdisplay, 500, 400
2 cgdisplay, 500, 400, wid=2
3 cgdisplay, 500, 400, /free
```

6.10 resizable windows

```
1 cgwindow, 'plot', findgen(11)
2
3 loadct, 33, rgb_table=palette
4 cgPlot, cgdemodata(1), layout=[2, 2, 3], color='red', /window
5 cgContour, cgdemodata(2), nlevels=12, layout=[2, 2, 1], $
6     xstyle=1, ystyle=1, color='dodger blue', /addcmd
7 cgSurf, cgDemoData(2), /elevation, layout=[2, 2, 4], $
8     palette=palette, /addcmd
9 cgImage, cgDemoData(19), multimargin=4, /axes, $
10    layout=[2, 2, 2], /addcmd
```




Figure 1: Resizable Window

7 Color

7.1 information about your graphic device

```

1 window
2 device, get_visual_name=theVisual, $
3         get_visual_depth=theDepth
4 print, theVisual, theDepth

```

Normally, you will get 'TrueColor' and '24'. **If not**, add following content to your
~/.Xdefaults

```

1 idl.gr_visual: TrueColor
2 idl.gr_depth: 24

```

Or, add following command to your IDL startup file

```
1 Device, TrueColor=24
```

7.2 color models

There are two color modes in IDL, i.e., *decomposed color* (default) and *indexed color*

- `1 Device, Decomposed=1 ; Selects Decomposed Color Model.`
- `2 Device, Decomposed=0 ; Selects Indexed Color Model.`

Every color in IDL is represented, ultimately, as a three-element byte vector of red, green, and blue values, in which each value can vary between 0 and 255. We call this vector a *color triple*. Totally, we can get $256^3 = 16777216$ kinds of different color. This is also known as *true color*.

In *decomposed* color model,

- `1 device, decomposed=1`
- `2 clr = [R,G,B]`
- `3 DecomposedColor = clr[0] + clr[1] * 2L ^ 8 + clr[2] * 2L ^ 16`




Figure 2: decomposed color, credit: David W. Fanning

Things differ in *indexed* color model. Suppose we load a at color index 100 with the `TVLCT` command, like this

- `1 device, decomposed=0`
- `2 TVLCT, R, G, B, 100`
- `3 IndexColor = 100`




Figure 3: indexed color, credit: David W. Fanning

Once a color index is loaded, the color palette is limited to 256, but it is enough in most of cases.

7.3 specify a color by name

use *cgColor* from the *Coyote Library*, or simply use *PickColorName* program

7.4 load color table indices

7.4.1 *TvLCT*

¹ *TvLCT*, red, green, blue, position

use *CIndex* command to see the result

7.4.2 *LoadCT*

used to load standard color tables come with IDL

7.4.3 XloadCT

7.4.4 Xpalette

8 Line plot

8.1 simple curve

```

1 curve=cgdemodata(1)
2 time = (findgen(100) + 1) * 6.0 / 100
3 plot,time,curve,xtitle='time axis',ytitle='signal strength',$ 
4      title='experiment 35M',charsize=1.5, color=cgcolor('black'), $ 
5      background=cgcolor('white')

```




Figure 4: simple plot

set default char size

```
1 !P.CharSize=1.5
```

8.2 OPlot

plot multiple curves on a single window

```

1 plot, curve, color=cgcolor('black'), background=cgcolor('white')
2 oplot, curve/2.0, linestyle=1, color=cgcolor('black')
3 oplot, curve/5.0, linestyle=2, color=cgcolor('black')

```




Figure 5: oplot

we can create second y axis by *save* keyword

```

1  plot, curve, ystyle=8, ytitle='solid line', $
2    position=[0.13,0.15,0.85,0.95], color=cgcolor('black'), $
3    background=cgcolor('white'), charsize=1.5
4  axis, yaxis=1, yrang=[0,max(curve*5+1)], /save, $
5    ytitle='dashed line', color=cgcolor('black'), charsize=1.5
6  oplot, curve*5, linestyle=2, color=cgcolor('black')
```




Figure 6: oplot two y axis

8.3 modify your plot

8.3.1 thick

A floating-point value between 1.0 (the default) and 10.0

8.3.2 line style

Table 21: LineStyle

Value of linestyle	LineStyle
0	solid line
1	dotted
2	dashed
3	dash dot
4	dash dot dot dot
5	long dash
6	no line draw

8.3.3 symbol

Table 22: Psym

Value of Psym	symbol
0	no symbol
1	plus sign
2	asterisk
3	period
4	diamond
5	triangle
6	square
7	X
8	UserSym
9	unused
10	ladder
-Psym	symbols connected by lines

We can create our user symbol

```

1  ;; define symbol
2  x = [0.0, 0.5, -0.8, 0.8, -0.5, 0.0]
3  y = [1.0, -0.8, 0.3, 0.3, -0.8, 1.0]
4  usersym, x, y, /fill
5  ;; plot
6  vector = randomu(seed, 10) * 10
7  plot, vector, color=cgcolor('black'), background=cgcolor('white'), $
8    xstyle=1
9  oplot, vector, color=cgcolor('red'), psym=8, $
10   symsize=4.0, noclip=1

```




Figure 7: plot user symbol

8.3.4 color

```
1 vector = randomu(seed, 10) * 10
2 plot, vector, color=cgcolor('black'), background=cgcolor('yellow'), $
3      /nodata
4 oplot, vector, color=cgcolor('red')
5 oplot, vector, color=cgcolor('forest green'), $
6      psym=2, symsize=3.0
```




Figure 8: plot color

8.3.5 range

```
1 plot, time, curve, xrange=[2,4], yrange=[-10,40]
```

8.3.6 [XYZ]Style

Table 23: [XYZ]Style

value of [xyz]style	result
1	accurate axis range
2	expanded axis range
4	no axis
8	no fram
16	Y axis don't start from 0

all are cumulative

```
1 plot, time, curve, xstyle=9, ystyle=9, color=cgcolor('black'), $  
2 background=cgcolor('white')
```




Figure 9: plot [xy]style

8.3.7 grid

when [xy]ticklen=1

```
1 plot, time, curve, xstyle=9, ystyle=1, xticklen=1, xgridstyle=1,$  
2 yticklen=1, ygridstyle=1, color=cgcolor('black'), $  
3 background=cgcolor('white')
```




Figure 10: plot grid

8.3.8 tick

1. when [xy]ticklen or ticklen is smaller than 0

```
1 plot, time, curve, xstyle=9, ystyle=9, xticklen=-0.03, $  
2      yticklen=-0.02, color=cgcolor('black'), $  
3      background=cgcolor('white')
```




Figure 11: tick1

2. use [xy]ticks and [xy]minor

the whole axis is divided into [xy]ticks parts and each part will be divided into [xy]minor parts

```

1 plot, time, curve, xstyle=1, xticks=2, xminor=10, yticks=3, $
2      yminor=5, xticklen=-0.03, yticklen=-0.02, $
3      color=cgcolor('black'), background=cgcolor('white')
```




Figure 12: tick2

8.3.9 axis labels

1. no axis labels

```

1 plot, time, vector, xtitle='Measurements', $
2      ytitle='Signal Strength', title='Experiment 1A', $
3      charsize=1.5, xcharsize=1.0, ycharsize=1.0, $
4      ystyle=1, xtickformat='(A1)', $
5      color=cgcolor('black'), background=cgcolor('white')
```

2. creating labels

```

1 time = findgen(6) * 15.0 / 5.0
2 vector = randomu(seed, 6) * 100
3 ticklabels = ['1st', '2nd', '3rd', '$
4      '4th', '5th', '6th']
5 plot, time, vector, xtitle='Measurements', $
6      ytitle='Signal Strength', title='Experiment 1A', $
7      charsize=1.5, xcharsize=1.0, ycharsize=1.0, $
8      xtickname=ticklabels, xticks=5, ystyle=1, $
9      color=cgcolor('black'), background=cgcolor('white')
```




Figure 13: Axis Labels

3. label format

```
1 plot, time, vector, xtitle='Elapsed Time', $  
2      ytitle='Signal Strength', title='Experiment 1A', $  
3      xtickformat='(F0.2)', ystyle=1, color=cgcolor('black'), $  
4      background=cgcolor('white'), charsize=1.5, $  
5      xcharsize=1.0, ycharsize=1.0
```




Figure 14: Label Format

4. calendar dates as axis labels

```

1  vector = randomu(seed, 6) * 10
2  time = timegen(n_elements(vector), $
3      start=julday(7, 1, 2016, 0, 0, 0), $
4      units='days')
5  void = label_date(date_format='%D %M %Y')
6  plot, time, vector, xtitle='Measurements', $
7      ytitle='Signal Strength', title='Experiment 1A', $
8      charsize=1.5, xcharsize=1.0, ycharsize=1.0, $
9      xtickformat='label_date', xticks=5, xstyle=1, $
10     color=cgcolor('black'), background=cgcolor('white'), $
11     position=[0.10, 0.15, 0.90, 0.85]

```




Figure 15: Calendar Labels

8.4 error bar plot

```

1  data = Congrid(cgDemoData(1), 15)
2  seed = -5L
3  time = cgScaleVector(Findgen(N_Elements(data)), 1, 9)
4  high_yerror = RandomU(seed, N_Elements(data)) * 5 > 0.5
5  low_yerror = RandomU(seed, N_Elements(data)) * 4 > 0.25
6  high_xerror = RandomU(seed, N_Elements(data)) * 0.75 > 0.1
7  low_xerror = RandomU(seed, N_Elements(data)) * 0.75 > 0.1
8
9  xtitle = 'Time'
10 ytitle = 'Signal Strength'
11 title = 'Error Bar Plot'
12 position = [0.125, 0.125, 0.9, 0.925]
13 thick = (!D.Name EQ 'PS') ? 3 : 1
14
15 cgDisplay, 600, 500, Title='Errorbar Plot'
16

```

```

17 cgPlot, time, data, Color='red5', PSym=-16, SymColor='olive', $  

18   SymSize=1.0, Thick=thick, Title=title, XTitle=xtitle, $  

19   YTitle=ytitle, Position=position, YRange=[-5, 35], $  

20   XRange=[0,10], YStyle=1, $  

21   ERR_XLow=low_xerror, ERR_XHigh=high_xerror, $  

22   ERR_YLow=low_yerror, ERR_YHigh=high_yerror, ERR_Color='blu5'

```




Figure 16: Error Bar Plot

8.5 histogram line plot

```

1 cgdisplay
2 image = cgdemodata(7)
3 h = histogram(image, binsize=5B, /nan, $  

4   omin=omin, omax=omax)
5 x = scale_vector(findgen(n_elements(h)), omin, omax)
6 plot, x, h, psym=10, max_value=15000, xstyle=1, $  

7   ytitle='Number of Pixels', xtitle='Pixel Value'

1 cghistoplot, cgdemodata(7), /fill, max_value=10000, $  

2   ytitle='Number of Pixels', xtitle='Pixel Value'

```




Figure 17: cgHistoplot

```

1 data = randomu(5L, 200) * 20
2 !p.multi = [0, 2, 2]
3 cghistoplot, data, binsize=1.0
4 cghistoplot, data, binsize=1.0, /fill, $
5      polycolor=['charcoal', 'dodger blue']
6 cghistoplot, data, binsize=1.0, /line_fill, $
7      polycolor=['charcoal', 'dodger blue'], $ 
8      orientation=[45, -45]
9 cghistoplot, data, binsize=1.0, /fill, $
10     polycolor='sky blue', mininput=0
11 moredata = randomu(3L, 80) * 20
12 cghistoplot, moredata, binsize=1.0, /fill, $
13     polycolor='royal blue', /oplot, mininput=0
14 !p.multi = 0

1 cghistoplot, cgdemodata(7), /fill, $
2      probcolorname='blu6', /oprobability

```




Figure 18: cgHistoplot Multiplot




Figure 19: cgHistoplot Cumulative Probability

8.6 missing data

```

1 data = float(cgdemodata(1))
2 data[40:49] = !values.f_nan
3 plot, data

```

8.7 cgplot

```

1 cgdisplay, 600, 300
2 cgplot, cgdemodata(1), psym=-16, color='red', $
    symcolor='blue', axiscolor='dark green'
3
1 cgplot, cgdemodata(1), aspect=2/3.0, title='Aspect Plot'
2
1 data = cgdemodata(1)
2 time = scale_vector(findgen(101), 0, 6)
3 cloadct, 33, /silent
4 cgplot, time, data, /nodata
5 cgplots, time, data, psym=-16, $
    color=bytscl(data), symcolor=bytscl(data), $
    symsize=scale_vector(data, 1.0, 2.5)
7

```




Figure 20: cgPlot

9 Contour

9.1 basic contour

```

1 peak = cgDemoData(2)
2 lat = FIndGen(41) * (24. / 40) + 24
3 lon = FindGen(41) * 50.0 / 40 - 122
4 Contour, peak, lon, lat, XTitle='Longitude', YTitle='Latitude', $
    xstyle=1, ystyle=1, /follow, color=cgcolor('black'), $
    background=cgcolor('white'), charsize=1.5, c_charsize=1.0
5
6

```




Figure 21: basic contour

9.2 levels

The *NLevels* keyword does not, generally, give you “N contouring levels” as claimed by the IDL documentation.

```

1 nlevels = 12
2 step = (max(peak) - min(peak)) / nlevels
3 vals = indgen(nlevels) * step + min(peak)
4 contour, peak, lon, lat, xstyle=1, ystyle=1, /follow, levels=vals, $
5      color=cgcolor('black'), background=cgcolor('white'), $
6      charsize=1.5, c_charsize=1.0

```




Figure 22: contour levels

9.3 labels

```
1 threeLevels = [250, 750, 1200]
2 annotations = ['Low', 'Medium', 'High']
3 contour, peak, lon, lat, xstyle=1, ystyle=1, /follow, $
4     levels = threeLevels, c_annotation=annotations, $
5     color=cgcolor('black'), background=cgcolor('white'), $
6     charsize=1.5, c_charsize=1.5
```




Figure 23: contour labels

9.4 modifying contour lines

9.4.1 *downhill*

We can put small tick marks in the downhill direction of the contour lines

```

1 nlevels = 12
2 step = (max(peak) - min(peak)) / nlevels
3 vals = indgen(nlevels) * step + min(peak)
4 contour, peak, lon, lat, xstyle=1, ystyle=1, /follow, $
5      levels=vals, /downhill, color=cgcolor('black'), $
6      background=cgcolor('white'), charsize=1.5, $
7      c_charsize=1.0

```




Figure 24: contour downhill

9.4.2 *c_linestyle*

```

1 contour, peak, lon, lat, xstyle=1, ystyle=1, /follow, $
2      levels=vals, C_linestyle=[0,2], color=cgcolor('black'), $
3      background=cgcolor('white'), charsize=1.5, c_charsize=1.0

```




Figure 25: contour linestyle

9.4.3 c_thick

```
1 contour, peak, lon, lat, xstyle=1, ystyle=1, /follow, $  
2 levels=vals, C_thick=[1,1,2], color=cgcolor('black'), $  
3 background=cgcolor('white'), charsize=1.5, c_charsize=1.0
```




Figure 26: contour thickness

9.4.4 line color

```
1 contour, peak, lon, lat, xstyle=1, ystyle=1, levels=vals, $  
2   color=cgcolor('blue'), c_color=cgcolor('orange'), $  
3   background=cgcolor('white'), /follow, charsize=1.5, $  
4   c_charsize=1.0
```




Figure 27: contour color

```
1 Device, Get_Decomposed=currentMode  
2 Device, Decomposed=0  
3 LoadCT, 33, NColors=12, Bottom=1  
4 Contour, peak, lon, lat, xstyle=1, ystyle=1, Levels=vals, /NoData, $  
5   Background=cgColor('white'), Color=cgColor('black'), $  
6   charsize=1.5, c_charsize=1.0  
7 Contour, peak, lon, lat, Levels=vals, /Overplot, $  
8   C_Colors=IndGen(12) + 1, /follow, charsize=1.5, $  
9   c_charsize=1.0  
10 Device, Decomposed=currentMode
```




Figure 28: contour more color

9.5 filling color

```
1 Device, Get_Decomposed=currentMode
2 Device, Decomposed=0
3 LoadCT, 33, NColors=12, Bottom=1
4 Contour, peak, lon, lat, xstyle=1, ystyle=1, Levels=vals, /fill, $
      C_Colors=IndGen(12)+1, Background=cgColor('white'), $
      Color=cgColor('black')
7 Device, Decomposed=currentMode
```




Figure 29: contour filling color

if you want to see contour lines on the color filled image, just use overplot

```
1 contour, peak, lon, lat, xstyle=1, ystyle=1, /follow, levels=vals, $  
2 /overplot, c_color=cgColor('white')
```




Figure 30: contour filling color overplot

Add a color bar

```

1 cgsetColorstate, 0, CurrentState=currentState
2 LoadCT, 33, NColors=nlevels, Bottom=1
3 contour, peak, lon, lat, /cell_fill, levels=vals, $
4     position=[0.125, 0.125, 0.95, 0.80], $
5     background=cgColor('white'), color=cgColor('black'), $
6     xstyle=1, ystyle=1, c_colors=indgen(nlevels)+1
7 contour, peak, lon, lat, /overplot, /follow, $
8     color=cgColor('black'), levels=vals
9 cgsetColorstate, currentState
10 cgColorbar, range=[min(peak), max(peak)], $
11     divisions=nlevels, ticklen=1, xminor=0, $
12     annotatecolor='black', ncolors=nlevels, bottom=1, $
13     position=[0.125, 0.915, 0.955, 0.95], charsize=0.75

```




Figure 31: contour color bar

9.6 cgContour

```

1 data = cgdemodata(2)
2 loadct, 0, /silent
3 window
4 cgcontour, data, title='Normal Contour Plot', $
5     xtitle='X Title', ytitle='Y Title'

```




Figure 32: cgContour Normal Plot

```
1 cloadct, 17, /brewer
2 cloadct, 4, ncolors=10, bottom=1, /brewer, /reverse
3 c_colors = indgen(10) + 1
4 position = [0.1, 0.1, 0.9, 0.8]
5 cgcontour, data, nlevels=10, /fill, $
6     position=position, c_colors=c_colors
7 cgcontour, data, nlevels=10, /overplot
8 cgcolorbar, divisions=10, ncolors=10, bottom=1, $
9     range=[min(data), max(data)], ticklen=1.0, $
10    position=[0.1, 0.90, 0.90, 0.94], charsize=0.75
```




Figure 33: cgContour Filled Contour

Table 24: cgContour Labels

<i>Label</i> keyword	Meaning
1 (default)	labels every contour level
2	labels every other contour
3	labels every third contour
0	no contour labeling

```

1 window, 0, xsize=400, ysize=400
2 cgcontour, data, nlevels=10, axiscolor='blue', $
  color='red'
3 window, 1, xsize=400, ysize=400
4 cgcontour, data, nlevels=5, axiscolor='brown', $
  c_colors=['aquamarine', 'dark green', 'orange', '$
    'crimson', 'purple']
5
6
7

```




Figure 34: cgContour Colors

10 Surface

10.1 basic

Surface of 2D data can be easily made using 'surface' command

```

1 peak=cgdemodata(2)
2 lon=findgen(41)*50./41-122
3 lat=findgen(41)*(24./40)+24
4 surface,peak,lon,lat,xtitle='longitude',$ 
    ytitle='latitude',ztitle='elevation', charsize=2.5, $
    color=cgcolor('black'), background=cgcolor('white'), $
    xstyle=1, ystyle=1, zstyle=0
5
6
7

```




Figure 35: surface

```
1 cgsurface,peak,lon,lat,xtitle='longitude',$  
2 ytitle='latitude',ztitle='elevation', $  
3 xstyle=1, ystyle=1, zstyle=0
```

10.2 title

We usually use *xyouts* to make title of the surface

```
1 xyouts,0.5,0.90,/normal,size=2.0,'Surface',align=0.5, $  
2 color=cgcolor('black')
```




Figure 36: xyouts

10.3 rotating

The default value of $A[xz]$ is 30

```
1 surface,peak,lon,lat,Ax=60,Az=60,charsize=2.5, $  
2 color=cgcolor('black'), background=cgcolor('white')
```




Figure 37: surface rotating

10.4 with contour

```
1 Surface, peak, lon, lat, charsize=2.5, $  
2   Color=cgColor('black'), $  
3   Background=cgColor('white'), $  
4   XStyle=5, YStyle=5, ZStyle=0, $  
5   AX=45, AZ=60, /Save  
6 Contour, peak, lon, lat, /T3D, XStyle=1, charsize=2.5, $  
7   YStyle=1, XTitle=xtitle, YTitle=ytitle, $  
8   /NoErase, Color=cgColor('black'), /NoData  
9 Contour, peak, lon, lat, /T3D, /Overplot, charsize=2.5, $  
10  NLevels=12, C_Label=Replicate(1,12), /NoClip, $  
11  ZValue=0.0, /Downhill, Color=cgColor('blu6')
```




Figure 38: Surface with Contour

10.5 skirt

```
1 surface, peak, lon, lat, charsize=2.5, $  
2   color=cgcolor('black'), $  
3   background=cgcolor('white'), $  
4   xstyle=1, ystyle=1, zstyle=0, $  
5   ax=30, az=60, skirt=0
```




Figure 39: Surface Skirt

10.6 color

```
1 surface, peak, lon, lat, charsize=2.5, $  
2     color=cgcolor('black'), $  
3     background=cgcolor('white'), $  
4     xstyle=1, ystyle=1, zstyle=0, /nodata  
5 surface, peak, lon, lat, charsize=2.5, $  
6     color=cgcolor('grn5'), /noerase, $  
7     xstyle=5, ystyle=5, zstyle=4
```




Figure 40: surface color

```
1 surface, peak, lon, lat, charsize=2.5, $  
2     color=cgcolor('black'), $  
3     background=cgcolor('white'), $  
4     xstyle=1, ystyle=1, zstyle=0  
5 surface, peak, lon, lat, charsize=2.5, $  
6     color=cgcolor('grn5'), /noerase, $  
7     xstyle=5, ystyle=5, zstyle=4, $  
8     bottom=cgcolor('red5')
```




Figure 41: Surface Bottom Color

```

1 surface, peak, lon, lat, charsize=2.5, $
2   color=cgcolor('black'), $
3   background=cgcolor('white'), $
4   xstyle=1, ystyle=1, zstyle=0
5 surface, peak, lon, lat, charsize=2.5, $
6   color=cgcolor('red5'), /noerase, $
7   xstyle=5, ystyle=5, zstyle=4, /skirt
8 surface, peak, lon, lat, charsize=2.5, $
9   color=cgcolor('grn5'), /noerase, $
10  xstyle=5, ystyle=5, zstyle=4

```




Figure 42: Surface Skirt Color

```

1 snowdepth = cgdemodata(3)
2 cgsetcolorstate, 0, currentstate=currentstate
3 set_shading, values=[0, 249]
4 loadct, 33
5 surface, peak, lon, lat, charsize=2.0, $
6   color=cgcolor('black', 250), $
7   background=cgcolor('white', 251), $
8   xstyle=1, ystyle=1, zstyle=0, $
9   shades=bytscl(snowdepth, top=249)
10 cgsetcolorstate, currentstate
11 cgcolorbar, ncolors=250, divisions=8, $
12   range=[min(peak), max(peak)], $
13   xminor=0, format='(I0)', $
14   annotatecolor='black', xtitle='', $
15   position=[0.2, 0.92, 0.8, 0.95], $
16   title='Pressure Units'

```




Figure 43: Surface Shading with a Second Data Set

10.7 shaded surface

```
1 shade_surf,peak,lon,lat,Az=45, charsize=2.0, $  
2           color=cgcolor('black'), $  
3           background=cgcolor('white')
```




Figure 44: shaded surface

```
1 cgsetcolorstate, 0, currentstate=currentstate  
2 set_shading, values=[0, 249], light=[0.0, 1.0, 0.5]
```

```

3 cgloadct, 5, /brewer, /reverse, ncolors=250
4 shade_surf, peak, lon, lat, charsize=2.0, $
5           color=cgcolor('black', 250), $
6           background=cgcolor('white', 251), $
7           xstyle=1, ystyle=1, zstyle=0
8 cgsetcolorstate, currentstate

```




Figure 45: shaded surface with color and different light position

```

1 cgsetcolorstate, 0, currentstate=currentstate
2 cgloadct, 5, /brewer, /reverse, ncolors=250
3 shade_surf, peak, lon, lat, charsize=2.0, $
4           color=cgcolor('black', 250), $
5           background=cgcolor('white', 251), $
6           xstyle=1, ystyle=1, zstyle=0, $
7           shades=bytscl(peak, top=249)
8 cgloadct, 33, ncolors=250
9 surface, peak, lon, lat, charsize=2.0, $
10          color=cgcolor('black', 250), $
11          background=cgcolor('white', 251), $
12          xstyle=5, ystyle=5, zstyle=4, /noerase, $
13          shades=bytscl(snowdepth, top=249)
14 cgsetcolorstate, currentstate

```




Figure 46: shaded surface with a wire mesh surface

10.8 3d scatter plot

```

1  ;; random data
2  data2d = cgdemodata(2)
3  s = size(data2d, /dimensions)
4  lon = (indgen(s[0]) * 25000L - 500000L) / 1000
5  lat = (indgen(s[1]) * 25000L - 500000L) / 1000
6  lat2d = rebin(reform(lat, 1, 41), 41, 41)
7  lon2d = rebin(lon, 41, 41)
8  pts = round(randomu(-3L, 200) * 41 * 41)
9  datairr = data2d[pts]
10 lonirr = lon2d[pts] + random(5L, 200) * 50 - 25
11 latirr = lat2d[pts] + random(8L, 200) * 50 - 25
12 surface, dist(100), xstyle=1, ystyle=1, /nodata, $
13     xrange=[-550, 550], yrange=[-550, 550], $
14     zrange=[0, 1500], color=cgcolor('black'), $
15     background=cgcolor('white'), charsize=2.0, $
16     position=[0.075, 0.075, 0.925, 0.750], /save
17  ;; add axes
18  axis, xaxis=1, /t3d, xstyle=1, charsize=2.0, $
19      color=cgcolor('black')
20  axis, yaxis=1, /t3d, ystyle=1, charsize=2.0, $
21      color=cgcolor('black')
22  axis, zaxis=0, /t3d, 550, 550, charsize=2.0, $
23      color=cgcolor('black')
24  axis, zaxis=0, /t3d, 550, -550, charsize=2.0, $
25      color=cgcolor('black')
26  ;; load color table
27  symcolors = fix(bytscl(datairr))
28  loadct, 33, /silent
29  ;; draw light gray lines
30  num = n_elements(datairr)
31  for j=0, num-1 do plots, [lonirr[j], lonirr[j]], $
```

```

32 [latirr[j], latirr[j]], $  

33 [datairr[j], 0], $  

34 color=cgcolor('light gray'), /t3d  

35 ;; draw a dot at the bottom of the light gray lines  

36 for j=0, num-1 do plots, lonirr[j], latirr[j], 0, $  

37 /t3d, psym=3, symsize=2.0, $  

38 color=cgcolor('black')  

39 ;; draw a filled circle symbol in color at the top of the  

40 ;; light gray lines  

41 for j=0, num-1 do plots, lonirr[j], latirr[j], $  

42 datairr[j], /t3d, psym=symcat(16), $  

43 symsize=4.0, $  

44 color=cgcolor(strtrim(symcolors[j], 2))  

45 ;; color bar  

46 loadct, 33  

47 cgcolorbar, divisions=6, format='(I0)', $  

48 annotatecolor='black', charsize=1.0, $  

49 position=[0.2, 0.88, 0.8, 0.92]

```




Figure 47: 3D Scatter Plot

10.9 cgSurf

```

1 cgsurf, cgdemodata(3), color='red', bottom='blue', $  

2 title='Colorful Surface', tsize=2.0

```

```

1 cgloadct, 5, /brewer, /reverse
2 cgsurf, data2d, lon, lat, charsize=2.0, $
3           xstyle=1, ystyle=1, zstyle=0, $
4           shades=bytscl(data2d), /shaded
5 cgloadct, 33
6 cgsurf, data2d, lon, lat, charsize=2.0, $
7           xstyle=5, ystyle=5, zstyle=4, /noerase, $
8           shades=bytscl(snowdepth)

1 cgloadct, 5, rgb_table=palette
2 cgsurf, cgdemodata(2), palette=palette, /elevation, $
3           charsize=2.0

1 cgwindow, wxsize=800, wysize=400
2 cgloadct, 33, /window
3 cgsurf, data2d, /elevation, layout=[2, 1, 1], $
4           /shaded, /addcmd
5 cgcontour, data2d, nlevels=12, /fill, $
6           layout=[2, 1, 2], /addcmd
7 cgcontour, data2d, nlevels=12, color='charcoal', $
8           layout=[2, 1, 2], /addcmd

```




Figure 48: cgSurf Multiplot

10.10 cgSurface

```

1 cgsurface, data2d, lon, lat, $
2           xtitle='Longitude', ytitle='Latitude', $
3           ztitle='Atmosphere Pressure'
4 cgsurface, data2d, lon, lat, $
5           xtitle='Longitude', ytitle='Latitude', $
6           ztitle='Atmosphere Pressure', /shaded

```




Figure 49: cgSurface Basic

```

1 earth = cgdemodata(7)
2 cgsurface, data2d, lon, lat, zscale=0.5, $
3           xtitle='Longitude', ytitle='Latitude', $
4           ztitle='Atmosphere Pressure', $
5           texture_image=earth, ctable=4, /brewer, $
6           title='2D Image as Texture'
7 rose = cgdemodata(16)
8 cgsurface, data2d, lon, lat, zscale=0.5, $
9           xtitle='Longitude', ytitle='Latitude', $
10          ztitle='Atmosphere Pressure', $
11          texture_image=rose, $
12          title='True-Color Image as Texture'
```




Figure 50: cgSurface Image as Texture

```

1 ;; read the image
2 file = filepath(subdir=['examples', 'data'], $ 
3                  'elev_t.jpg')
4 read_jpeg, file, image
5 ;; read the 64x64 dem byte data
6 file = filepath(subdir=['examples', 'data'], $
```

```

7           'elevbin.dat')
8 dem = bytarr(64, 64)
9 openr, lun, file, /get_lun
10 readu, lun, dem
11 free_lun, lun
12 ;; place the image onto the dem as a texture map
13 cgsurface, dem, texture_image=image, zscale=0.5

```




Figure 51: cgSurface 3D Map

11 Image position

11.1 margin

Margins are controlled by *!x.margin* and *!y.margin*, also by keywords of plot command *xmargin* and *ymargin*. The default values are [10,3] and [3,3], which means N times of character width of margin each side.

11.2 relative position

Relative position is controlled by *!p.position* or keyword *position* of plot command, which means the relative position of the left bottom corner and the top right corner. The value

is between 0 and 1.

```
1 contour, peak, position=[0.1, 0.1, 0.5, 0.9]
```

11.3 region

Similar to relative position, position of region is controlled by *p.region*, but no plot keyword available. Don't forget to reset the system variable to 0 after use.

11.4 multiple image

P.Multi is used to create multi-image in a window or PostScript page.

Table 25: !P.Multi

item	meaning
0	number of images rest you want to draw
1	number of columns
2	number of rows
3	number of floors in z axis
4	by row=0, or by column=1

Don't forget to reset it to 0 after use.

Sometimes we have to add a title, hence the out margin needs to be modified.

```
1 !p.Multi = [0,2,2,0,1]
2 !y.0margin = [2,4]
3 Plot, cgDemoData(1), color=cgcolor('black'), $
4     background=cgcolor('white')
5 Contour, cgDemoData(2), xstyle=1, ystyle=1, nlevels=10, $
6     color=cgcolor('black')
7 Surface, cgDemoData(2), color=cgcolor('black')
8 Shade_Surf, cgDemoData(2), color=cgcolor('black'), $
9     background=cgcolor('white')
10 XYOuts, 0.5, 0.9, /Normal, 'Four Graphics Plots', align=0.5, $
11     charsize=2.5, color=cgcolor('black')
12 !p.Multi = 0
```




Figure 52: multi image

Another way is to use *Coyote library*,

```
1 cgdisplay
2 cgloadct, 33, rgb_table=palette
3 cgplot, cgdemodata(1), layout=[2, 2, 3], color='red'
4 cgcontour, cgdemodata(2), nlevels=12, $
5     layout=[2, 2, 1], color='dodger blue'
6 cgsurf, cgdemodata(2), /elevation, layout=[2, 2, 4], $
7     palette=palette
8 cgimage, cgdemodata(19), multimargin=4, /axes, $
9     layout=[2, 2, 2]
```




Figure 53: multi image

12 Add text to image

12.1 font type

Font is determined by `!P.Font` or `Plot` keyword `Font`.

Table 26: `!P.Font`

<code>!P.Font</code>	Font Type
-1	Vector font (aka soft font or Hershey font)
0	hard font
1	TrueType font

Set default *TrueType* font to *Courier*,

```
device, set_font='Courier', /tt_font
```

The same as *Times*, *Helvetica* and *Symbol*.

12.2 XYOutS

12.2.1 position

The first parameter means X, the second parameter means Y by data coord by default. We can add *Normal* keyword to use normalized position

```

1 time = findgen(100)*6.0/100.0
2 plot, time, cgDemoData(1), Position=[0.15, 0.15, 0.95, 0.85]
3 XYOutS, 0.2, 0.92, 'Results: Experiment 35F3a', Size=2.0, /Normal

```

12.2.2 vector font

Table 27: Hershey font and index in IDL

value	font	value	font
!3	Simplex Roman	!12	Simplex Script
!4	Simplex Greek	!13	Complex Script
!5	Duplex Roman	!14	Gothic Italian
!6	Complex Roman	!15	Gothic German
!7	Complex Greek	!16	Cyrillic
!8	Complex Italian	!17	Triplex Roman
!9	Math Font	!18	Triplex Italian
!10	Special Characters	!20	Miscellaneous
!11	Gothic English	!X	(Back to default)

we can use *showfont* command to view each font

```

1 window, /free, xsize=600, ysize=600
2 showfont, 9, ''

```

use different vector font,

```

1 time = findgen(100)*6.0/100.0
2 plot, time, cgDemoData(1), Position=[0.15, 0.15, 0.95, 0.85], $
3      xtitle='Time', ytitle='Signal', color=cgcolor('black'), $
4      background=cgcolor('white'), charsize=1.5
5 XYOutS, 0.2, 0.92, '!12Results: Experiment 35F3a!X', Size=2.0, $
6      /Normal, color=cgcolor('black')

```




Figure 54: another vector font

12.2.3 alignment

Table 28: Alignment

Value	Alignment
0.0	left
0.5	center
1.0	right

```

1 plot, time, cgDemoData(1), Position=[0.15, 0.15, 0.95, 0.85], $
2      xtitle='Time', ytitle='Signal', color=cgcolor('black'), $
3      background=cgcolor('white'), charsize=1.5
4 XYOutS, 0.5, 0.92, '!12Results: Experiment 35F3a!X', Size=2.0, $
5      /Normal, align=0.5, color=cgcolor('black')
```

will align center.

12.2.4 delete

we can delete text by repeat the previous `XYOutS` command followed by `Color!=P.Background` or other background color used.

12.2.5 orientation

Using the keyword `orientation` (between -180 and 180).

13 Image I/O of common formats

13.1 formats

Table 29: image formats

Format	Read	Write	Query
BMP	read_bmp	write_bmp	query_bmp
GIF	read_gif	write_gif	query_gif
JPEG	read_jpeg	write_jpeg	query_jpeg
PICT	read_pict	write_pict	query_pict
PBM/PPM	read_ppm	write_ppm	query_ppm
PNG	read_png	write_png	query_png

13.2 reading image data

```

1 startPath = File_DirName(Filepath(Root_Dir=!Dir, $           Subdirectory=['examples', 'data'], '*'))
2
3 file = Dialog_Pickfile(File='muscle.jpg', $                  Filter='*.jpg', Path=startPath)
4
5 Read_JPEG, file, image

```

or you can use *read_image*

```
1 image = read_image(file)
```

for *png* and *tiff* format image

```

1 image = read_png(png_file)
2 image = read_image(png_file)
3 image = read_tiff(file, r, g, b, GeoTIFF=geo)
4 image = read_image(file, r, g, b, GeoTIFF=geo)

```

for *gif* file

```
1 read_gif, file, image, r, g, b
```

13.3 information about image files

```

1 validFile = query_image(file, info)
2 help, info, /structure
3 print, info.dimensions

```

There is a program in the *Coyote Library* named *ImageSelect* that takes advantage of *Dialog_Pickfile*, and the *Read_** and *Query_** routines to allow you to select common image files to read.

```
1 image = imageselect(filter='*.jpg', /demo)
```

13.4 writing image files

let's take a snapshot and write the result to the image file

```
1 cghistoplot, cgdemodata(7), /fill
2 snapshot = tvrd(true=1)
3 help, snapshot
```

the resulting image can be sent to any appropriate output file type.

```
1 write_jpeg, 'test.jpg', snapshot, true=1, quality=75
2 write_png, 'test.png', snapshot
3 write_tiff, 'test.tiff', reverse(snapshot)
4 write_image, 'test.jpg', 'JPEG', snapshot
```

Or you can simply use *cgSnapshot*

```
1 void = cgsnapshot(filename='test', /jpeg)
2 void = cgsnapshot(filename='test', /png)
3 void = cgsnapshot(filename='test', /tiff)
4 void = cgsnapshot(filename='test', /jpeg, /nodialog)
```

14 Display image

14.1 traditional image display commands

14.1.1 TVScl

It is not recommended to use this command. You'd better use *BytScl* to byte scale an image data and then display the image

14.1.2 TV

Not recommended either :-(

If you want to use the *TV* command to display a 2D image, then you need to use the indexed color model when you display the image.

display true color image

```
1 rose = cgDemoData(16)
2 tv, rose, true=1
```

Unfortunately The *TV* command wouldn't display an alpha channel image properly in any case

Basically, it takes on the order of about 25 lines of code to display an image properly with the *TV* command. And that is only if you want to get the colors right. If you want the image to show up in the right place in a PostScript file, or if you want to do other smart things, and believe me you do, you can multiply this number by a factor of 10 or so to get the number of commands you really need. – *David W. Fanning*

14.2 alternative image display commands

14.2.1 *ImDisp (by Liam Gumley)*

<http://www.gumley.com/PIP/Programs/imdisp.pro>

14.2.2 *cgImage (Coyote Library)*

14.3 positioning

```
1 position = [0.15,0.15,0.9,0.9]
2 stern = (cgdemodata(10))[*,* ,1]
3 cgimage, stern, position=position
```



Figure 55: cgImage

```
1 position = [0.15,0.15,0.9,0.9]
2 stern = (cgdemodata(10))[*,* ,1]
3 cgimage, stern, position=position, /keep_aspect
```



Figure 56: cgImage keep aspect

```
1 cgImage, stern, Margin=0.1, /Keep_Aspect, /Axes, $  
2      XRange=[-10,10], YRange=[0,50], Color='navy', $  
3      AxKeywords={XTicklen:-0.035, YTickLen:-0.035}
```




Figure 57: cgImage coordinate

14.4 erase graphics window before display

```
1 cgErase, 'white'  
2 cgImage, stern, margin=0.1
```

14.5 image backgrounds

```

1 cgImage, stern, background='wheat', margin=0.15, /axes, $
2           color='opposite'
```

by setting the axes color to *opposite*, /cgColor/ will calculate a color that contrast with the background color

14.6 display images with colors

```

1 cgLoadCT, 13, /brewer, /reverse, rgb_table=palette
2 cgImage, stern, background='wheat', margin=0.15, $
3           /axes, color='opposite', palette=palette
```




Figure 58: cgImage color

14.7 combining images with other graphical display

```

1 image = cgDemoData(18)
2 cgLoadCT, 4, /brewer, /reverse
3 cgImage, image, /save, /axes, /keep_aspect
4 cgContour, image, nlevels=10, label=2, /overplot
```




Figure 59: cgImage color

14.8 interacting with images

use *cgimageinfo* to identify the location of the cursor in the display and returns the value in that place, interactively

```

1 cgimageinfo,image
2 Click cursor in image window (window 0) to get image values.
3 Use LEFT mouse button to inquire.
4 Use RIGHT mouse button to exit program.
5
6 Expecting cursor clicks in image window...
7 Value at (85,258) is 7
8 Value at (143,141) is 213
9 Value at (176,169) is 255
10 Value at (303,327) is 8
11 Value at (175,384) is 25
12
13 Good-bye. cgImageInfo has returned control to IDL.

```

14.9 identify locations in the image

```

1 Cursor, xloc, yloc, /Down, /Device
2 normalcoords = convert_coord(xloc, yloc, /device, /to_normal)
3 xn = normalcoords[0]
4 yn = normalcoords[1]
5 p = position
6 dims=image_dimensions(image, xsize=xsize, ysize=ysize)
7 xvec = scale_vector(findgen(xsize + 1), p[0], p[2])
8 yvec = scale_vector(findgen(ysize + 1), p[1], p[3])
9 xpixel = value_locate(xvec, xn)
10 ypixel = value_locate(yvec, yn)
11 print, image[xpixel, ypixel]

```

14.10 zooming images

14.10.1 for non-true-color images

```

1 minimg = min(stern, max=maximg)
2 subregion = stern[50:99, 50:99]
3 window, xscale=400, yscale=400
4 cgimage, bytscl(subregion, min=minimg, max=maximg)

```

14.10.2 for true-color-images

```

1 rose = cgdemodata(16)
2 subregion = rose[:, 50:100, 50:100]
3 cgimage, subregion

```

14.10.3 FSC_ZImage

an interactive zoom image program from *Coyote Library*

14.11 multiple images

```

1 head = cgdemodata(8)
2 head = reverse(head, 2)
3 s=size(head, /dimensions)
4 cloadct, 0
5 cgdisplay, 9*s[0], 6*s[1]
6 !p.multi = [0, 9, 6]
7 !x.omargin = [0, 25]
8 !y.omargin = [0, 8]
9 for j=0, 53 do cgimage, head[:, :, j], $
10                                background='white', multimargin=0.25
11 !p.multi = 0
12 cgColorbar, /vertical, range=[0, 10], $
13                                position=[0.94, 0.15, 0.97, 0.80], $
14                                format='(F0.2)'
15 cgtext, 0.45, 0.95, 'MRI Study of the Head', $
16                                /normal, Alignment=0.5
17 !x.omargin = 0
18 !y.omargin = 0

```




Figure 60: cgImage Multi-display

The MultiMargin keyword is a four-element array that specifies margins about the bottom, left, top, and right of the image, respectively.

```
1 loadct, 33, /silent
2 !p.multi = [0, 2, 2]
3 margin = [4, 4, 2, 0]
4 cgimage, cgdemodata(18), /keep_aspect, /axes, $
5     multimargin=margin
6 cgimage, cgdemodata(18), /keep_aspect, /axes, $
7     multimargin=margin
8 cgimage, cgdemodata(18), /keep_aspect, /axes, $
9     multimargin=margin
10 cgimage, cgdemodata(18), /keep_aspect, /axes, $
11     multimargin=margin
12 !p.multi = 0
```




Figure 61: cgImage MultiMargin

14.12 transparent images

Transparent images have an alpha channel that indicates how the transparent image is to be “blended” with the information already on the display. Alpha channels are generally coded as values from 0 to 1 (or, in images, from 0 to 255), with 0 indicating total transparency (i.e., display the background pixel) and 1 indicating total opacity (i.e., display the foreground pixel).

14.12.1 image with alpha channel

download a png file from <http://www.idlcoyote.com/books/tg/data/toucan.png>

```

1 toucan = read_png('toucan.png')
2 !p.multi = [0, 2, 1]
3 window, xsize=162*2, ysize=150
4 cgimage, toucan[0:2, *, *]
5 cgimage, reform(toucan[3, *, *])
6 !p.multi = 0

```




Figure 62: cgImage separate alpha

```
1 window, xsize=400, ysize=400
2 cgerase, color='bisque'
3 cgimage, toucan, /keep_aspect, /noerase, margin=0.2
```




Figure 63: cgImage alpha channel image

```
1 window, 1, xsize=400, ysize=400
2 marsfile = Filepath(Subdir=['examples', 'data'], $
3                         'marsglobe.jpg')
4 read_jpeg, marsfile, mars
5 cgimage, mars
6 cgimage, toucan, alphafgpos=[0.3, 0.3, 0.7, 0.7]
```




Figure 64: cgImage alpha channel image with background image

14.12.2 creating blended images

```
1 scanfile = Filepath(Subdir=['examples','data'], $  
2                         'md1107g8a.jpg')  
3 Read_JPEG, scanfile, scan  
4 marsfile = Filepath(Subdir=['examples', 'data'], $  
5                         'marsglobe.jpg')  
6 read_jpeg,marsfile,mars  
7 window,xsize=400,ysize=400  
8 LoadCT, 0  
9 cgImage, scan, /NoInterp  
10 snapshot = cgSnapshot()  
11 alpha = 0.6  
12 blended = (snapshot * alpha) + (mars * (1-alpha))  
13 Window, XSize=400*3, YSize=400  
14 !P.Multi = [0,3,1]  
15 cgImage, mars  
16 cgImage, scan  
17 cgImage, blended  
18 !P.Multi = 0
```




Figure 65: cgImage blended image

or you can use *cgBlendImage* program from *Coyote Library*

```

1 window,xsize=400,ysize=400
2 cgBlendImage,snapshot,mars,alpha=0.6

```

14.12.3 creating transparent images

```

1 logfile = Filepath(Subdir=['examples','data'], $
2   'examples.tif')
3 logo = Read_TIFF(logofile, r, g, b)
4 logo = Reverse(logo, 2)
5 TVLCT, r, g, b
6 Window, XSize=375, YSize=150
7 cgImage, logo, /NoInterp
8 logo24 = cgSnapshot()
9 alpha = bytarr(375,150)+255B
10 logo24 = Transpose(logo24, [1, 2, 0])
11 red = logo24[*, *, 0]
12 grn = logo24[*, *, 1]
13 blu = logo24[*, *, 2]
14 blackIndices = Where( (red EQ 0) AND (grn EQ 0) $ 
15   AND (blu EQ 0), count)
16 IF count GT 0 THEN alpha[blackIndices] = 0B
17 logo32 = [[red]], [[grn]], [[blu]], [[alpha]] ]
18 logo32 = Transpose(logo32, [2,0,1])
19 Write_PNG, 'idl_logo.png', logo32
20 image = Make_Transparent_Image(Color='black', $
21   Filename='idl_logo.png', $
22   /Save_PNG)
23 Window, XSize=400, YSize=400
24 cgImage, mars
25 cgImage, logo32, alphafgpos=[0.30, 0.05, 0.70, 0.25]

```



Figure 66: idl logo



Figure 67: cgImage transparent logo

14.12.4 *transparent*

```
1 cgDisplay, 500, 500
2 cgImage, cgDemoData(7), CTIndex=0
3 cgImage, cgDemoData(5), CTIndex=33, Transparent=30, $
4           Missing_Value=0, AlphaFGPos=[0.5, 0.5, 1.0, 1.0]
```




Figure 68: cgImage transparent display

14.13 image in resizable graphics windows

14.13.1 *ImDisp*

```
1 image = cgdemodata(19)
2 cgwindow, 'imdisp', image, margin=0.2, /erase
```

14.13.2 *cgImage*

```
1 cgimage, image, margin=0.2, /window
```

when combined with other graphical display

```
1 cgplot, cgdemodata(1), yrange=[0, 80], /window
2 cgimage, image, position=[0.7, 0.5, 0.9, 0.9], $
    /keep_aspect, /addcmd
```




Figure 69: cgImage transparent display

15 Postscript output

15.1 how does it work

```

1 thisDevice = !d.name
2 set_plot, 'ps'
3 device, filename='plot_example.ps'
4 x = findgen(200) * 0.1
5 plot, x, sin(x)
6 device, /close_file
7 set_plot, thisDevice

```

The `!d.name` system variable stores the name of the current graphics device. This is `WIN` on a Windows operating system, and is `X` on an Unix operating system. Once you have changed your graphics device, any graphical output will be sent to the new graphics device, e.g., when `!d.name` is set to `PS`, graphical output will be written into a PostScript file. By default, the PostScript file that is created is named `idl.ps`, which can be modified with the `filename` keyword to the `Device` command.

15.2 change the configuration

You can get the information of the current state of the PostScript device by

```

1 set_plot, 'ps'
2 help, /device

```

The default page orientation is *Portrait* with color output disabled. If you switch to *Landscape* orientation and check the configuration

```
1 device, /landscape
2 help, /device
```

Table 30: Commonly used *device* keywords for the *PS* device

Keyword	Purpose
filename	Set the output file name (default: idl.ps)
/portrait	Select portrait orientation (default)
/landscape	Select landscape orientation
/color	Select color output (default if grayscale)
bits_per_pixel	Set number of bits per pixel (default is 4)
xsize	Set width of drawable area
ysize	Set height of drawable area
xoffset	Set offset of drawable area origin from left of page
yoffset	Set offset of drawable area origin from bottom of page
/inches	Set units of inches for size and offset keywords in this call to device only (default is centimeters)
font_size	Set font size (points, default is 12)
/close_file	Complete and close the output file

15.3 producing color (or grayscale) output

15.3.1 24-bit color

```
1 thisdevice = !d.name
2 set_plot, 'ps'
3 device, /color, bits_per_pixel=8
4 set_plot, thisdevice
```

Setting the *color* keyword automatically copies the current color table vectors into the PostScript file, although colors can be changed at any time in PostScript files by loading colors and color tables. You can also copy the current color table vectors into the PostScript file at the time you set the PostScript device by using the *copy* keyword to *set_plot*.

```
1 set_plot, 'ps', /copy
```

But the input color mode now is *index* by default, which makes the device act like an 8-bit index color graphics device. To put the PostScript device into a 24-bit color mode, you set the *decomposed* keyword to 1.

```
1 device, decomposed=1
```

or use the routines from the *Coyote Library*

```
1 cgsetcolorstate, 1, currentstate=currentstate
```

15.3.2 color background

On a display monitor, which is a raster device,

```
1 plot, cgdemodata(1), background=cgcolor('wheat'), $  
2     color=cgcolor('opposite')
```

However, this same command would result in a black on a white background if sent to a PostScript file, you have to 'fill' the PostScript file first.

```
1 thisdevice = !d.name  
2 set_plot, 'ps'  
3 device, filename='plot_fill_back.ps'  
4 device, /color, bits_per_pixel=8  
5 cgsetcolorstate, 1, currentstate=currentstate  
6 polyfill, [0,1,1,0,0], [0,0,1,1,0], /normal, $  
    color=cgcolor('wheat')  
8 plot, cgdemodata(1), color=cgcolor('opposite'), $  
    /noerase  
10 device, /close_file  
11 set_plot, thisdevice
```

or, simply use *cgplot*

```
1 thisdevice = !d.name  
2 set_plot, 'ps'  
3 device, filename='plot_fill_back.ps'  
4 device, /color, bits_per_pixel=8  
5 cgsetcolorstate, 1, currentstate=currentstate  
6 cgplot, cgdemodata(1), background='wheat'  
7 device, /close_file  
8 set_plot, thisdevice
```

15.4 window aspect ratio

The *Coyote Library* contains a function, named *pswindow*, that can "match" the aspect ratio of the current graphics window with a centered PostScript window having the same aspect ratio.

```
1 window, xsize=600, ysize=400  
2 output = pswindow(/metric)  
3 help, output, /structure
```

then simply add the *output* as extra arguments

```
1 set_plot, 'ps'  
2 device, _extra=output
```

We can also use *pswindow* command to set the window aspect ratio rather than matching the window.

```
1 output = pswindow(aspectratio=2./3)
```

15.5 setting size and offset manually




Figure 70: Portrait and Landscape orientation

Now if you'd like to reproduce the plot in Portrait orientation, on a 8.5- inch \times 11-inch page.

```

1  ;; Set size and offset
2  page_width = 8.5
3  page_height = 11.0
4  xsize = 6.5
5  ysize = 5.5
6  xoffset = (page_width - xsize) * 0.5
7  yoffset = (page_height - ysize) * 0.5
8  ;; Config the PostScript device and plot
9  this_device = !d.name
10 set_plot, 'ps'
11 device, filename='plot_port.ps', /portrait
12 device, xsize=xsize, ysize=ysize, $
    xoffset=xoffset, yoffset=yoffset, /inches
13 x = findgen(200) * 0.1
14 plot, x, sin(x)
15 device, /close_file
16 set_plot, this_device

```

To create the same plot in Landscape orientation with the same page size, the offset of the drawable area must be recomputed using a method that takes account of the new orientation

```

1  ;; Set offset for landscape orientation
2  xoffset = (page_width - ysize) * 0.5
3  yoffset = (page_height - xsize) * 0.5 + xsize
4  ;; Config and plot
5  this_device = !d.name
6  set_plot, 'ps'
7  device, filename='plot_landscape.ps', /landscape
8  device, xsize=xsize, ysize=ysize, $
    xoffset=xoffset, yoffset=yoffset, /inches
9  x = findgen(200) * 0.1

```

```

11 plot, x, sin(x)
12 device, /close_file
13 set_plot, this_device

```

However, the Landscape file produced is **UPSIDE DOWN**! You can simply use the *Cgfixps* routine from the *Coyote Library* to fix the bug

```
1 cgfixps, 'plot_landscape.ps'
```

15.6 setting size and offset automatically

```

1 PRO PSON, FILENAME=FILENAME, PAPER=PAPER, MARGIN=MARGIN, $
2 PAGE_SIZE=PAGE_SIZE, INCHES=INCHES, ASPECT=ASPECT, $
3 LANDSCAPE=LANDSCAPE, QUIET=QUIET
4
5 ;; Check arguments
6 if (n_elements(filename) eq 0) then filename = 'idl.ps'
7 if (n_elements(paper) eq 0) then paper = 'LETTER'
8 if (n_elements(margin) eq 0) then begin
9     margin = 2.5
10    endif else begin
11        if keyword_set(inches) then margin = margin * 2.54
12    endelse
13
14 ;; Check if PostScript mode is active
15 if !d.name eq 'PS' then begin
16     message, 'POSTSCRIPT output is already active', /continue
17     return
18 endif
19
20 ;; Get ratio of character width/height to screen width/height
21 xratio = float(!d.x_ch_size) / float(!d.x_vsize)
22 yratio = float(!d.y_ch_size) / float(!d.y_vsize)
23
24 ;; Save current device information in common block
25 common pson_infomation, info
26 info = {device:!d.name, window:!d.window, font:!p.font, $
27         filename:filename, xratio:xratio, yratio:yratio}
28
29 ;; Get size of page (centimeters)
30 widths = [[8.5, 8.5, 11.0, 7.25] * 2.54, 21.0, 29.7]
31 heights = [[11.0, 14.0, 17.0, 10.50] * 2.54, 29.7, 42.0]
32 names = ['LETTER', 'LEGAL', 'TABLOID', 'EXECUTIVE', $
33           'A4', 'A3']
34 index = where(strupcase(paper) eq names, count)
35 if (count ne 1) then begin
36     message, 'PAPER selection not supported', /continue
37     return
38 endif
39 page_width = widths[index[0]]
40 page_height = heights[index[0]]
41
42 ;; If page size was supplied, use it
43 if (n_elements(page_size) eq 2) then begin
44     page_width = page_size[0]
45     page_height = page_size[1]
46     if keyword_set(inches) then begin

```

```

47     page_width = page_width * 2.54
48     page_height = page_height * 2.54
49   endif
50 endif
51
52 ;; Compute aspect ratio of page when margins are subtracted
53 page_aspect = float(page_height - 2.0 * margin) / $
54           float(page_width - 2.0 * margin)
55
56 ;; Get aspect ratio of current graphics window
57 if (!d.window ge 0) then begin
58   win_aspect = float(!d.y_vsize) / float(!d.x_vsize)
59 else begin
60   win_aspect = 512.0 / 640.0
61 endelse
62
63 ;; If aspect ratio was supplied, use it
64 if (n_elements(aspect) eq 1) then $
65   win_aspect = float(aspect)
66
67 ;; Compute size of drawable area
68 case keyword_set(landscape) of
69   0 : begin
70     if (win_aspect ge page_aspect) then begin
71       ysize = page_height - 2.0 * margin
72       xsize = ysize / win_aspect
73     endif else begin
74       xsize = page_width - 2.0 * margin
75       ysize = xsize * win_aspect
76     endelse
77   end
78   1 : begin
79     if (win_aspect ge (1.0 / page_aspect)) then begin
80       ysize = page_width - 2.0 * margin
81       xsize = ysize / win_aspect
82     endif else begin
83       xsize = page_height - 2.0 * margin
84       ysize = xsize * win_aspect
85     endelse
86   end
87 endcase
88
89 ;; Compute offset of drawable area from page edges
90 if (keyword_set(landscape) eq 0) then begin
91   xoffset = (page_width - xsize) * 0.5
92   yoffset = (page_height - ysize) * 0.5
93 else begin
94   xoffset = (page_width - ysize) * 0.5
95   yoffset = (page_height - xsize) * 0.5 + xsize
96 endelse
97
98 ;; Switch to PostScript device
99 set_plot, 'PS'
100 device, landscape=keyword_set(landscape), scale_factor=1.0
101 device, xsize=xsize, ysize=ysize, $
102           xoffset=xoffset, yoffset=yoffset
103 device, filename=filename, /color, bits_per_pixel=8
104

```

```

105    ;; Set character size
106    xcharsize = round(info.xratio * !d.x_vsize)
107    ycharsize = round(info.yratio * !d.y_vsize)
108    device, set_character_size=[xcharsize, ycharsize]
109
110    ;; Report to user
111    if (keyword_set(quiet) eq 0) then $
112        print, filename, $
113            format='("Started POSTSCRIPT output to ", a)'
114
115 END

1 PRO PSOFF, QUIET=QUIET
2
3    ;; Check that PostScript output is active
4    if (!d.name ne 'PS') then begin
5        message, 'POSTSCRIPT output not active: ' + $
6            'nothing done', /continue
7        return
8    endif
9
10   ;; Get entry device information from common block
11   common pson_infomation, info
12   if (n_elements(info) eq 0) then begin
13       message, 'PSON was not called prior to PSOFF: ' + $
14           'nothing done', /continue
15       return
16   end
17
18   ;; Close PostScript device
19   device, /close_file
20   set_plot, info.device
21
22   ;; Restore window and font
23   if (info.window ge 0) then wset, info.window
24   !p.font = info.font
25
26   ;; Report to user
27   if (keyword_set(quiet) eq 0) then $
28       print, info.filename, $
29           format='("Ended POSTSCRIPT output to ", a)'
30
31 END

1 pson, file='surface.ps', margin=1.0
2 device, color=0, bits_per_pixel=4
3 z = dist(64)
4 z = shift(z, 32, 32)
5 z = exp(-(z * 0.1) ^ 2)
6 shade_surf, z, xstyle=1, ystyle=1, charsize=2, $
7 pixels=1000, xmargin=[4, 1], ymargin=[1, 0]
8 psoff

```




Figure 71: Surface

15.7 encapsulated postscript output

To produce PostScript output that can be inserted directly into other Post- Script documents, such as journal articles or books, you must select the encapsulated option before you send graphics output to the PostScript file.

```
1 device, /encapsulated
```

If you would like to be able to see a representation of the graphic in the document into which you imported the file, you must specify a value for the *preview* keyword. A Preview keyword value of 1 will cause the Post- Script driver to include a bit-mapped image of the graphic along with its PostScript description. If you would prefer to use a TIFF preview image, instead of a bit-mapped image, set the *preview* keyword to 2.

```
1 device, /encapsulated, preview=2
```

15.8 setting the language level

```
1 device, language_level=2
```

15.9 using PostScript fonts

15.9.1 font system

The font type is controlled by `.p.font` variable. A value of -1 selects Hershey fonts, a value of 0 selects hardware fonts, and a value of 1 selects TrueType fonts.

```

1 this_device, !d.name
2 set_plot, 'ps'
3 device, _extra=pswindow(aspectratio=1.0/3), font_size=12, $
4     /encapsulated, filename='three_font_system.ps', $
5     /helvetica, /landscape
6 !p.multi=[0, 3, 1]
7 data = cgdemodata(1)
8 cplot, data, font=-1, title='Hershey Fonts'
9 cplot, data, font=0, title='Hardware Fonts'
10 cplot, data, font=1, title='TrueType Fonts'
11 device, /close_file
12 !p.multi=0
13 set_plot, this_device

```

15.9.2 font mapping

Table 31: Default mapping of fonts

Number	Hershey Font	PostScript Font
!3	Simplex Roman	Helvetica
!4	Simplex Greek	Helvetica-Bold
!5	Duplex Roman	Helvetica-Narrow
!6	Complex Roman	Helvetica-Narrow-BoldOblique
!7	Complex Greek	Times-Roman
!8	Complex Italian	Times-BoldItalic
!9	Math Font	Symbol
!10	Special Characters	ZapfDingbats
!11	Gothic English	Courier
!12	Simplex Script	Courier-Oblique
!13	Complex Script	Palatino-Roman
!14	Gothic Italian	Palatino-Italic
!15	Gothic German	Palatino-BoldItalic
!16	Cyrillic	AvantGrande-Book
!17	Triplex Roman	NewCenturySchlbk-Roman
!18	Triplex Italian	NewCenturySchlbk-Bold
!20	Miscellaneous	Undefined
!X	(Back to default)	(Back to default)

```

1 thisdevice = !d.name
2 set_plot, 'ps'
3 device, /zapfchancery, font_size=9, /encapsulated, filename='zaph.ps'
4 cplot, cgdemodata(1), font=0, title='ZapfChancery Font', $
5     xtitle='Time', ytitle='Signal Strength'
6 device, /close_file
7 set_plot, thisdevice

```




Figure 72: ZapfChancery Font

15.9.3 TrueType Fonts

```

1 set_plot, 'ps'
2 device, set_font='Times', /tt_font
3 !p.font = 1

```

for Bold and Italic fonts

```

1 device, set_font='Times Bold', /tt_font
2 device, set_font='Times Italic', /tt_font
3 device, set_font='Times Bold Italic', /tt_font

```

15.9.4 rotating fonts

```

1 set_plot, 'x'
2 cgplot, cgdemodata(1), font=0, xtitle='Good Fonts Here', $
    ytitle='Bad Fonts Here'
3
1 set_plot, 'ps'
2 device, _extra=pswindow(aspectratio=1.0/2)
3 !p.multi = [0, 2, 1]
4 cgplot, cgdemodata(1), font=0, $
    xtitle='Good Fonts', ytitle='And Good Fonts'
5 cgsurf, cgdemodata(2), xtitle='X Fonts', $
    ytitle='Y Fonts', ztitle='Z Fonts', font=0
6 !p.multi=0
7 device, /close_file
8 cfixps, 'idl.ps'
9 set_plot, 'x'

```

```

1 set_plot, 'ps'
2 device, _extra=pswindow(aspectratio=1.0/2)
3 !p.multi = [0, 2, 1, 0, 1]
4 cgsurf, cgdemodata(2), xtitle='X Fonts', $ 
5     ytitle='Y Fonts', ztitle='Z Fonts', $ 
6     Font=-1, title='Hershey Fonts'
7 cgsurf, cgdemodata(2), xtitle='X Fonts', $ 
8     ytitle='Y Fonts', ztitle='Z Fonts', $ 
9     Font=1, title='TrueType Fonts'
10 !p.multi = 0
11 device, /close_file
12 cfixps, 'idl.ps'
13 set_plot, 'x'
```

15.9.5 Greek characters

1. using embedded font

```

1 window, /free, xsize=600, ysize=600
2 showfont, 9, 'Symbol'

1 thisdevice = !d.name
2 set_plot, 'ps'
3 mu = '!9' + String("155B") + '!X'
4 cgplot, cgdemodata(1), font=1, $ 
5     xtitle='Wavelength (' + mu + 'm)', charsize=1.0
6 device, /close_file
7 set_plot, thisdevice
```

The *Coyote Library* contains a program named *cgGreek* makes it easier for us to use Greek font. You can select lowercase Greek letters by specifying the Greek letter wanted in lowercase letters. You can select uppercase Greek letters either by setting the Capital keyword or by making the first letter of the Greek letter name a capital letter.

```
1 void = cggreek(/example)
```

2. using *textoidl*

You need to install the TeXtoIDL program.

```

1 showtex

1 cgps_open, 'textoidl.ps'
2 cgdisplay, 500, 100
3 astring = textoidl('\Sigma_{n=1}^{\infty} 1/n^2 = \pi^2/6')
4 xyouts, 0.5, 0.5, align=0.5, charsize=3.0, astring
5 cgps_close, /png, /width=500
```

$$\sum_{n=1}^{\infty} \frac{1}{n^2} = \frac{\pi^2}{6}$$

Figure 73: TeXtoIDL

3. using *latexify*

See Slug's guide.

```

1 pson, file='latexify.eps', margin=1.0
2 cgplot, findgen(10), xtitle='xt', charsize=1.5, font=0
3 psoff
4 latexify, 'latexify.eps', 'xt', 'Mass [M_{\odot}]'

```




Figure 74: *latexify*

15.9.6 configuring the PostScript device interactively

```

1 keywords = psconfig()
2 thisdevice = !d.name
3 set_plot, 'ps'
4 device, _extra=keywords

```

If *psconfig* is called with the *match* keyword set, the initial window configuration will match the aspect ratio of the current graphics window.

```

1 cgdisplay, 600, 300
2 keywords = psconfig(/match)
3 help, keywords, /structure

```

For font information

```

1 keywords = psconfig(/fontinfo)

```

```

1 thisdevice = !d.name
2 thisfont = !p.font
3 keywords = psconfig(cancel=cancel, /fontinfo, fonttype=1)
4 if cancel then return
5 !p.font = keywords.fonttype
6 set_plot, 'ps'
7 ;; graphics commands here
8 device, /close_file
9 set_plot, thisdevice
10 !p.font = thisfont

```

If you prefer not to use the graphical user interface to *psconfig*,

```

1 set_plot, 'ps'
2 device, _extra=psconfig(filename='coyote.ps', /match, /nogui)

```

15.9.7 display images in PostScript

```

1 thisdevice = !d.name
2 set_plot, 'ps'
3 image = congrid(cgdemodata(7), 400, 400)
4 device, xsize=5, ysize=5, /inches, decomposed=1
5 tv, image, 0.5, 0.5, xsize=4, ysize=4, /inches
6 cgplot, findgen(100), /nodata, /noerase, $
    position=[0.1, 0.1, 0.9, 0.9], font=0, charsize=1
7 device, /close_file
8 set_plot, thisdevice

1 rose = cgdemodata(16)
2 set_plot, 'ps'
3 image2d = color_quan(rose, 1, r, g, b)
4 tvlct, r, g, b
5 tv, image2d
6 device, /close_file

```